Ansatz Schwarzschild metric

Based on Arindam Kumar Chatterjee Video Lecture 4 The Cartan formalism Application to spherically symmetric spacetimes

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2)
```
info:build_info()$info@version;
```

(%o2)

5.38.1

(%i2)
```
reset()$kill(all)$
```

(%i1)
```
derivabbrev:true$
```

(%i2)
```
ratprint:false$
```

(%i3)
```
fpprintprec:5$
```

(%i4)
```
if get('itensor,'version)=false then load(itensor)$
```

(%i5)
```
if get('ctensor,'version)=false then load(ctensor)$
```

(%i11)
```
ctrgsimp:true$
ratchristof:true$
ratriemann:true$
rateinstein:true$
ratweyl:true$
ratfac:true$
```

(%i12)
```
declare(trigsimp,evfun)$
```

1 Coordinate metric

```
(%i14) iframe_flag:false$
       cframe_flag:false$
```

```
(%i18) assume(0≤r)$
       assume(0≤θ,θ≤π)$
       assume(0≤sin(θ))$
       assume(0≤ϕ,ϕ≤2*π)$
```

```
(%i19) init_ctensor()$
```

```
(%i20) ξ:ct_coords:[t,r,θ,ϕ]$
```

```
(%i21) dim:length(ct_coords)$
```

Line Element

```
(%i22) depends([A,B,C],r)$
```

```
(%i23) assume(A>0,B>0,C>0)$
```

```
(%i24) ldisplay(ds^2=line_element:A^2*del(t)^2-B^2*del(r)^2-C^2*del(θ)^2-C^2*sin(θ)^2*del(ϕ)^2)$
```

$$ds^2 = -C^2 \sin(\theta)^2 \operatorname{del}(\phi)^2 - C^2 \operatorname{del}(\theta)^2 + A^2 \operatorname{del}(t)^2 - B^2 \operatorname{del}(r)^2 \tag{\%t24}$$

Covariant Metric Tensor

```
(%i28) lg:zeromatrix(dim,dim)$
       for i thru dim do
       lg[i,i]:factor(coeff(expand(line_element),del(ξ[i])^2))$
       for j thru dim do for k thru dim do
       if j≠k then lg[j,k]:factor(expand(ratsimp(1/2*coeff(coeff(expand(line_element),del(ξ[j])),del(ξ[k]
       ishow(g([μ,ν],[])=lg)$
```

$$g_{\mu\nu} = \begin{pmatrix} A^2 & 0 & 0 & 0 \\ 0 & -B^2 & 0 & 0 \\ 0 & 0 & -C^2 & 0 \\ 0 & 0 & 0 & -C^2 \sin(\theta)^2 \end{pmatrix} \tag{\%t28}$$

Sets up the package for further calculations

```
(%i29) cmetric()$
```

Contravariant Metric Tensor

```
(%i30) ishow(g([],[μ,ν])=ug:invert(lg))$
```

$$g^{\mu\nu} = \begin{pmatrix} \frac{1}{A^2} & 0 & 0 & 0 \\ 0 & -\frac{1}{B^2} & 0 & 0 \\ 0 & 0 & -\frac{1}{C^2} & 0 \\ 0 & 0 & 0 & -\frac{1}{C^2 \sin(\theta)^2} \end{pmatrix} \tag{\%t30}$$

Line element

(%i31) `ldisplay(ds^2=diff(ξ).lg.transpose(diff(ξ)))$`

$$ds^2 = -C^2 \sin(\theta)^2 \operatorname{del}(\phi)^2 - C^2 \operatorname{del}(\theta)^2 + A^2 \operatorname{del}(t)^2 - B^2 \operatorname{del}(r)^2 \tag{\%t31}$$

Christoffel Symbol of the first kind

(%i33)
```
christof(false)$
for i thru dim do for j:i thru dim do for k thru dim do
if lcs[i,j,k]≠0 then
ishow(Γ([ξ[i],ξ[j],ξ[k]],[])=lcs[i,j,k])$
```

$$\Gamma_{ttr} = -A\,(A_r) \tag{\%t33}$$

$$\Gamma_{trt} = A\,(A_r) \tag{\%t33}$$

$$\Gamma_{rrr} = -B\,(B_r) \tag{\%t33}$$

$$\Gamma_{r\theta\theta} = -C\,(C_r) \tag{\%t33}$$

$$\Gamma_{r\phi\phi} = -C\,(C_r)\,\sin(\theta)^2 \tag{\%t33}$$

$$\Gamma_{\theta\theta r} = C\,(C_r) \tag{\%t33}$$

$$\Gamma_{\theta\phi\phi} = -C^2\,\cos(\theta)\,\sin(\theta) \tag{\%t33}$$

$$\Gamma_{\phi\phi r} = C\,(C_r)\,\sin(\theta)^2 \tag{\%t33}$$

$$\Gamma_{\phi\phi\theta} = C^2\,\cos(\theta)\,\sin(\theta) \tag{\%t33}$$

Christoffel Symbol of the second kind

(%i35)
```
christof(false)$
for i thru dim do for j:i thru dim do for k thru dim do
if mcs[i,j,k]≠0 then
ishow(Γ([ξ[i],ξ[j]],[ξ[k]])=mcs[i,j,k])$
```

$$\Gamma^{r}_{tt} = \frac{A\,(A_r)}{B^2} \tag{\%t35}$$

$$\Gamma^{t}_{tr} = \frac{A_r}{A} \tag{\%t35}$$

$$\Gamma^{r}_{rr} = \frac{B_r}{B} \tag{\%t35}$$

$$\Gamma^{\theta}_{r\theta} = \frac{C_r}{C} \tag{\%t35}$$

$$\Gamma^{\phi}_{r\phi} = \frac{C_r}{C} \tag{\%t35}$$

$$\Gamma^{r}_{\theta\theta} = -\frac{C\,(C_r)}{B^2} \tag{\%t35}$$

$$\Gamma^{\phi}_{\theta\phi} = \frac{\cos(\theta)}{\sin(\theta)} \tag{\%t35}$$

$$\Gamma^{r}_{\phi\phi} = -\frac{C\ (C_r)\ \sin(\theta)^2}{B^2} \tag{\%t35}$$

$$\Gamma^{\theta}_{\phi\phi} = -\cos(\theta)\ \sin(\theta) \tag{\%t35}$$

Riemann Tensor

(%i39)
```
riemann(false)$
lriemann(false)$
uriemann(false)$
for a thru dim do for b thru dim do
for c thru (if symmetricp(lg,dim) then b else dim) do
for d thru (if symmetricp(lg,dim) then a else dim) do
if riem[a,b,c,d]≠0 then
ishow(R([ξ[a],ξ[b],ξ[c]],[ξ[d]])=riem[a,b,c,d])$
```

$$R^{t}_{rrt} = \frac{(A_r)\ (B_r) - (A_{rr})\ B}{AB} \tag{\%t39}$$

$$R^{t}_{\theta\theta t} = -\frac{(A_r)\ C\ (C_r)}{A\ B^2} \tag{\%t39}$$

$$R^{r}_{\theta\theta r} = -\frac{C\ (B\ (C_{rr}) - (B_r)\ (C_r))}{B^3} \tag{\%t39}$$

$$R^{t}_{\phi\phi t} = -\frac{(A_r)\ C\ (C_r)\ \sin(\theta)^2}{A\ B^2} \tag{\%t39}$$

$$R^{r}_{\phi\phi r} = -\frac{C\ (B\ (C_{rr}) - (B_r)\ (C_r))\ \sin(\theta)^2}{B^3} \tag{\%t39}$$

$$R^{\theta}_{\phi\phi\theta} = -\frac{(C_r - B)\ (C_r + B)\ \sin(\theta)^2}{B^2} \tag{\%t39}$$

(%i40)
```
for a thru dim do for b thru dim do
for c thru (if symmetricp(lg,dim) then b else dim) do
for d thru (if symmetricp(lg,dim) then a else dim) do
if lriem[a,b,c,d]≠0 then
ishow(R([ξ[d],ξ[a],ξ[b],ξ[c]],[])=lriem[a,b,c,d])$
```

$$R_{trrt} = \frac{A\ ((A_r)\ (B_r) - (A_{rr})\ B)}{B} \tag{\%t40}$$

$$R_{t\theta\theta t} = -\frac{A\ (A_r)\ C\ (C_r)}{B^2} \tag{\%t40}$$

$$R_{r\theta\theta r} = \frac{C\ (B\ (C_{rr}) - (B_r)\ (C_r))}{B} \tag{\%t40}$$

$$R_{t\phi\phi t} = -\frac{A\ (A_r)\ C\ (C_r)\ \sin(\theta)^2}{B^2} \tag{\%t40}$$

$$R_{r\phi\phi r} = \frac{C\ (B\ (C_{rr}) - (B_r)\ (C_r))\ \sin(\theta)^2}{B} \tag{\%t40}$$

$$R_{\theta\phi\phi\theta} = \frac{C^2\ (C_r - B)\ (C_r + B)\ \sin(\theta)^2}{B^2} \tag{\%t40}$$

Ricci Tensor

(%i44)
```
ric:zeromatrix(dim,dim)$
ricci(false)$
uricci(false)$
for i thru dim do for j:i thru dim do
if ric[i,j]≠0 then
ishow(R([ξ[i],ξ[j]])=ric[i,j])$
```

$$R_{tt} = \frac{A\,(2\,(A_r)\,B\,(C_r) - (A_r)\,(B_r)\,C + (A_{rr})\,BC)}{B^3 C} \tag{\%t44}$$

$$R_{rr} = -\frac{2AB\,(C_{rr}) - 2A\,(B_r)\,(C_r) - (A_r)\,(B_r)\,C + (A_{rr})\,BC}{ABC} \tag{\%t44}$$

$$R_{\theta\theta} = -\frac{ABC\,(C_{rr}) + AB\,(C_r)^2 - A\,(B_r)\,C\,(C_r) + (A_r)\,BC\,(C_r) - A\,B^3}{A\,B^3} \tag{\%t44}$$

$$R_{\phi\phi} = -\frac{\left(ABC\,(C_{rr}) + AB\,(C_r)^2 - A\,(B_r)\,C\,(C_r) + (A_r)\,BC\,(C_r) - A\,B^3\right)\sin(\theta)^2}{A\,B^3} \tag{\%t44}$$

Returns a list of the unique differential equations

(%i45)
```
map(ldisp,efe:findde(ric,2))$
```

$$2\,(A_r)\,B\,(C_r) - (A_r)\,(B_r)\,C + (A_{rr})\,BC \tag{\%t45}$$

$$2AB\,(C_{rr}) - 2A\,(B_r)\,(C_r) - (A_r)\,(B_r)\,C + (A_{rr})\,BC \tag{\%t46}$$

$$ABC\,(C_{rr}) + AB\,(C_r)^2 - A\,(B_r)\,C\,(C_r) + (A_r)\,BC\,(C_r) - A\,B^3 \tag{\%t47}$$

(%i48)
```
deindex;
```

$$[[1,1],[2,2],[3,3]] \tag{\%o48}$$

Scalar curvature

(%i49)
```
scurvature();
```

$$(2(2ABC\,(C_{rr}) + AB\,(C_r)^2 - 2A\,(B_r)\,C\,(C_r) + 2\,(A_r)\,BC\,(C_r) - (A_r)\,(B_r)\,C^2 + (A_{rr})\,B\,C^2 - A\,B^3))/(A\,B^3\,C^2) \tag{\%o49}$$

Kretschmann invariant

(%i50)
```
rinvariant();
```

$$\frac{8(B\,(C_{rr}) - (B_r)\,(C_r))^2}{B^6\,C^2} + \frac{4(C_r - B)^2\,(C_r + B)^2}{B^4\,C^4} + \frac{8(A_r)^2\,(C_r)^2}{A^2\,B^4\,C^2} + \frac{4((A_r)\,(B_r) - (A_{rr})\,B)^2}{A^2\,B^6} \tag{\%o50}$$

Einstein Tensor

(%i54)
```
ein:zeromatrix(dim,dim)$
einstein(false)$
leinstein(false)$
for i thru dim do for j:i thru dim do
if lein[i,j]≠0 then
ishow(G([ξ[i]],[ξ[j]])=ein[i,j])$
```

$$G_t^t = -\frac{2BC\,(C_{rr}) + B\,(C_r)^2 - 2\,(B_r)\,C\,(C_r) - B^3}{B^3\,C^2} \quad (\%t54)$$

$$G_r^r = -\frac{A\,(C_r)^2 + 2\,(A_r)\,C\,(C_r) - A\,B^2}{A\,B^2\,C^2} \quad (\%t54)$$

$$G_\theta^\theta = -\frac{AB\,(C_{rr}) - A\,(B_r)\,(C_r) + (A_r)\,B\,(C_r) - (A_r)\,(B_r)\,C + (A_{rr})\,BC}{A\,B^3C} \quad (\%t54)$$

$$G_\phi^\phi = -\frac{AB\,(C_{rr}) - A\,(B_r)\,(C_r) + (A_r)\,B\,(C_r) - (A_r)\,(B_r)\,C + (A_{rr})\,BC}{A\,B^3C} \quad (\%t54)$$

Returns a list of the unique differential equations

(%i55)
```
map(ldisp,efe:findde(ein,2))$
```

$$2BC\,(C_{rr}) + B\,(C_r)^2 - 2\,(B_r)\,C\,(C_r) - B^3 \quad (\%t55)$$

$$A\,(C_r)^2 + 2\,(A_r)\,C\,(C_r) - A\,B^2 \quad (\%t56)$$

$$AB\,(C_{rr}) - A\,(B_r)\,(C_r) + (A_r)\,B\,(C_r) - (A_r)\,(B_r)\,C + (A_{rr})\,BC \quad (\%t57)$$

(%i58)
```
deindex;
```

$$[[1,1],[2,2],[3,3]] \quad (\%o58)$$

Clean up

(%i62)
```
forget(0≤r)$
forget(0≤θ,θ≤π)$
forget(0≤sin(θ))$
forget(0≤ϕ,ϕ≤2*π)$
```

(%i63)
```
elapsed_real_time();
```

$$13.11 \quad (\%o63)$$

(%i64)
```
elapsed_run_time();
```

$$2.297 \quad (\%o64)$$

2 Tetrad metric

Causes computations to be performed relative to a moving frame as opposed to a holonomic metric

(%i65)
```
kill(labels)$
```

(%i2)
```
iframe_flag:true$
cframe_flag:true$
```

(%i6)
```
assume(0≤r)$
assume(0≤θ,θ≤π)$
assume(0≤sin(θ))$
assume(0≤ϕ,ϕ≤2*π)$
```

(%i7)
```
init_ctensor()$
```

(%i8)
```
ξ:ct_coords:[t,r,θ,ϕ]$
```

(%i9)
```
dim:length(ct_coords)$
```

Line Element

(%i10)
```
depends([A,B,C],r)$
```

(%i11)
```
assume(A>0,B>0,C>0)$
```

(%i12)
```
ldisplay(ds^2=line_element:A^2*del(t)^2-B^2*del(r)^2-C^2*del(θ)^2-C^2*sin(θ)^2*del(ϕ)^2)$
```

$$ds^2 = -C^2 \sin(\theta)^2 \operatorname{del}(\phi)^2 - C^2 \operatorname{del}(\theta)^2 + A^2 \operatorname{del}(t)^2 - B^2 \operatorname{del}(r)^2 \qquad (\%t12)$$

If cframe_flag is true, the function expects that the values of fri (the inverse frame matrix) and lfg (the frame metric) are defined. From these, the frame matrix fr and the inverse frame metric ufg are computed.

The covariant frame metric
The covariant frame metric lfg (background metric)

(%i13)
```
lfg:matrix([1,0,0,0],[0,-1,0,0],[0,0,-1,0],[0,0,0,-1]);
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix} \qquad (\mathrm{lfg})$$

The frame field inverse stored as a matrix
The inverse frame matrix fri (coframe) (coframe covectors)

(%i14)
```
fri:matrix([A,0,0,0],[0,B,0,0],[0,0,C,0],[0,0,0,C*sin(θ)]);
```

$$\begin{pmatrix} A & 0 & 0 & 0 \\ 0 & B & 0 & 0 \\ 0 & 0 & C & 0 \\ 0 & 0 & 0 & C \sin(\theta) \end{pmatrix} \qquad (\mathrm{fri})$$

Sets up the package for further calculations

(%i15)
```
cmetric(false)$
```

Covariant metric tensor

(%i16) `ishow(g([μ,ν],[])=lg)$`

$$g_{\mu\nu} = \begin{pmatrix} A^2 & 0 & 0 & 0 \\ 0 & -B^2 & 0 & 0 \\ 0 & 0 & -C^2 & 0 \\ 0 & 0 & 0 & -C^2 \sin(\theta)^2 \end{pmatrix} \tag{\%t16}$$

The inverse frame metric
The inverse frame metric ufg

(%i17) `ldisplay(ufg)$`

$$ufg = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix} \tag{\%t17}$$

The frame field stored as a matrix
The frame matrix fr (frame vectors)

(%i18) `ldisplay(fr)$`

$$fr = \begin{pmatrix} \frac{1}{A} & 0 & 0 & 0 \\ 0 & -\frac{1}{B} & 0 & 0 \\ 0 & 0 & -\frac{1}{C} & 0 \\ 0 & 0 & 0 & -\frac{1}{C \sin(\theta)} \end{pmatrix} \tag{\%t18}$$

Contravariant metric tensor

(%i19) `ishow(g([],[μ,ν])=ug:invert(lg))$`

$$g^{\mu\nu} = \begin{pmatrix} \frac{1}{A^2} & 0 & 0 & 0 \\ 0 & -\frac{1}{B^2} & 0 & 0 \\ 0 & 0 & -\frac{1}{C^2} & 0 \\ 0 & 0 & 0 & -\frac{1}{C^2 \sin(\theta)^2} \end{pmatrix} \tag{\%t19}$$

Line element

(%i20) `ldisplay(ds^2=diff(ξ).lg.transpose(diff(ξ)))$`

$$ds^2 = -C^2 \sin(\theta)^2 \operatorname{del}(\phi)^2 - C^2 \operatorname{del}(\theta)^2 + A^2 \operatorname{del}(t)^2 - B^2 \operatorname{del}(r)^2 \tag{\%t20}$$

Christoffel Symbol of the first kind

(%i22)
```
christof(false)$
for i thru dim do for j:i thru dim do for k thru dim do
if lcs[i,j,k]≠0 then
ishow('Γ([ξ[i],ξ[j],ξ[k]],[])=lcs[i,j,k])$
```

$$\Gamma_{ttr} = -\frac{A_r}{AB} \tag{\%t22}$$

$$\mathit{\Gamma}_{trt} = \frac{A_r}{AB} \qquad (\%t22)$$

$$\mathit{\Gamma}_{\theta\theta r} = \frac{C_r}{BC} \qquad (\%t22)$$

$$\mathit{\Gamma}_{\phi\phi r} = \frac{C_r}{BC} \qquad (\%t22)$$

$$\mathit{\Gamma}_{\phi\phi\theta} = \frac{\cos(\theta)}{C\,\sin(\theta)} \qquad (\%t22)$$

Christoffel Symbol of the second kind

(%i24)
```
christof(false)$
for i thru dim do for j:i thru dim do for k thru dim do
if mcs[i,j,k]≠0 then
ishow('Γ([ξ[i],ξ[j]],[ξ[k]])=mcs[i,j,k])$
```

$$\mathit{\Gamma}_{tt}^{r} = \frac{A_r}{AB} \qquad (\%t24)$$

$$\mathit{\Gamma}_{tr}^{t} = \frac{A_r}{AB} \qquad (\%t24)$$

$$\mathit{\Gamma}_{\theta\theta}^{r} = -\frac{C_r}{BC} \qquad (\%t24)$$

$$\mathit{\Gamma}_{\phi\phi}^{r} = -\frac{C_r}{BC} \qquad (\%t24)$$

$$\mathit{\Gamma}_{\phi\phi}^{\theta} = -\frac{\cos(\theta)}{C\,\sin(\theta)} \qquad (\%t24)$$

Riemann Tensor all up

(%i26)
```
uriemann(false)$
for a thru dim do for b thru dim do
for c thru (if symmetricp(lg,dim) then b else dim) do
for d thru (if symmetricp(lg,dim) then a else dim) do
if uriem[a,b,c,d]≠0 then
ishow('R([],[ξ[a],ξ[b],ξ[c],ξ[d]])=uriem[a,b,c,d])$
```

$$R^{rrtt} = \frac{(A_r)\,(B_r) - (A_{rr})\,B}{A\,B^3} \qquad (\%t26)$$

$$R^{\theta\theta tt} = -\frac{(A_r)\,(C_r)}{A\,B^2 C} \qquad (\%t26)$$

$$R^{\theta\theta rr} = \frac{B\,(C_{rr}) - (B_r)\,(C_r)}{B^3 C} \qquad (\%t26)$$

$$R^{\phi\phi tt} = -\frac{(A_r)\,(C_r)}{A\,B^2 C} \qquad (\%t26)$$

$$R^{\phi\phi rr} = \frac{B\,(C_{rr}) - (B_r)\,(C_r)}{B^3 C} \qquad (\%t26)$$

$$R^{\phi\phi\theta\theta} = \frac{(C_r - B)\,(C_r + B)}{B^2\,C^2} \qquad (\%t26)$$

Riemann Tensor

(%i28)
```
riemann(false)$
for a thru dim do for b thru dim do
for c thru (if symmetricp(lg,dim) then b else dim) do
for d thru (if symmetricp(lg,dim) then a else dim) do
if riem[a,b,c,d]≠0 then
ishow('R([" ",ξ[b],ξ[c],ξ[d]],[ξ[a]])=riem[a,b,c,d])$
```

$$R^{r}_{rtt} = \frac{(A_r)\,(B_r) - (A_{rr})\,B}{A\,B^3} \tag{\%t28}$$

$$R^{\theta}_{\theta tt} = -\frac{(A_r)\,(C_r)}{A\,B^2 C} \tag{\%t28}$$

$$R^{\theta}_{\theta rr} = -\frac{B\,(C_{rr}) - (B_r)\,(C_r)}{B^3 C} \tag{\%t28}$$

$$R^{\phi}_{\phi tt} = -\frac{(A_r)\,(C_r)}{A\,B^2 C} \tag{\%t28}$$

$$R^{\phi}_{\phi rr} = -\frac{B\,(C_{rr}) - (B_r)\,(C_r)}{B^3 C} \tag{\%t28}$$

$$R^{\phi}_{\phi\theta\theta} = -\frac{(C_r - B)\,(C_r + B)}{B^2\,C^2} \tag{\%t28}$$

Riemann Tensor all down

(%i30)
```
lriemann(false)$
for a thru dim do for b thru dim do
for c thru (if symmetricp(lg,dim) then b else dim) do
for d thru (if symmetricp(lg,dim) then a else dim) do
if lriem[a,b,c,d]≠0 then
ishow('R([ξ[d],ξ[a],ξ[b],ξ[c]],[])=lriem[a,b,c,d])$
```

$$R_{trrt} = \frac{(A_r)\,(B_r) - (A_{rr})\,B}{A\,B^3} \tag{\%t30}$$

$$R_{t\theta\theta t} = -\frac{(A_r)\,(C_r)}{A\,B^2 C} \tag{\%t30}$$

$$R_{r\theta\theta r} = \frac{B\,(C_{rr}) - (B_r)\,(C_r)}{B^3 C} \tag{\%t30}$$

$$R_{t\phi\phi t} = -\frac{(A_r)\,(C_r)}{A\,B^2 C} \tag{\%t30}$$

$$R_{r\phi\phi r} = \frac{B\,(C_{rr}) - (B_r)\,(C_r)}{B^3 C} \tag{\%t30}$$

$$R_{\theta\phi\phi\theta} = \frac{(C_r - B)\,(C_r + B)}{B^2\,C^2} \tag{\%t30}$$

Ricci Tensor

(%i33)
```
ric:zeromatrix(dim,dim)$
ricci(false)$
for i thru dim do for j:i thru dim do
if ric[i,j]≠0 then
ishow('R([ξ[i],ξ[j]])=ric[i,j])$
```

$$R_{tt} = \frac{2\,(A_r)\,B\,(C_r) + ((A_{rr})\,B - (A_r)\,(B_r))\,C}{A\,B^3 C} \quad (\%t33)$$

$$R_{rr} = -\frac{2AB\,(C_{rr}) - 2A\,(B_r)\,(C_r) + ((A_{rr})\,B - (A_r)\,(B_r))\,C}{A\,B^3 C} \quad (\%t33)$$

$$R_{\theta\theta} = -\frac{ABC\,(C_{rr}) + AB\,(C_r)^2 + ((A_r)\,B - A\,(B_r))\,C\,(C_r) - A\,B^3}{A\,B^3\,C^2} \quad (\%t33)$$

$$R_{\phi\phi} = -\frac{ABC\,(C_{rr}) + AB\,(C_r)^2 + ((A_r)\,B - A\,(B_r))\,C\,(C_r) - A\,B^3}{A\,B^3\,C^2} \quad (\%t33)$$

Ricci Tensor all up

(%i36)
```
uric:zeromatrix(dim,dim)$
uricci(false)$
for i thru dim do for j:i thru dim do
if uric[i,j]≠0 then
ishow('R([],[ξ[i],ξ[j]])=uric[i,j])$
```

$$R^{tt} = \frac{2\,(A_r)\,B\,(C_r) - (A_r)\,(B_r)\,C + (A_{rr})\,BC}{A\,B^3 C} \quad (\%t36)$$

$$R^{rr} = \frac{2AB\,(C_{rr}) - 2A\,(B_r)\,(C_r) - (A_r)\,(B_r)\,C + (A_{rr})\,BC}{A\,B^3 C} \quad (\%t36)$$

$$R^{\theta\theta} = \frac{ABC\,(C_{rr}) + AB\,(C_r)^2 - A\,(B_r)\,C\,(C_r) + (A_r)\,BC\,(C_r) - A\,B^3}{A\,B^3\,C^2} \quad (\%t36)$$

$$R^{\phi\phi} = \frac{ABC\,(C_{rr}) + AB\,(C_r)^2 - A\,(B_r)\,C\,(C_r) + (A_r)\,BC\,(C_r) - A\,B^3}{A\,B^3\,C^2} \quad (\%t36)$$

Returns a list of the unique differential equations

(%i37)
```
map(ldisp,efe:findde(ric,2))$
```

$$2\,(A_r)\,B\,(C_r) - (A_r)\,(B_r)\,C + (A_{rr})\,BC \quad (\%t37)$$

$$2AB\,(C_{rr}) - 2A\,(B_r)\,(C_r) - (A_r)\,(B_r)\,C + (A_{rr})\,BC \quad (\%t38)$$

$$ABC\,(C_{rr}) + AB\,(C_r)^2 - A\,(B_r)\,C\,(C_r) + (A_r)\,BC\,(C_r) - A\,B^3 \quad (\%t39)$$

(%i40)
```
deindex;
```

$$[[1,1],[2,2],[3,3]] \quad (\%o40)$$

Scalar curvature

(%i41) `scurvature();`

$$(2(2ABC\ (C_{rr})+AB\,(C_r)^2-2A\ (B_r)\,C\ (C_r)+2\,(A_r)\,BC\ (C_r)-(A_r)\ (B_r)\ C^2+(A_{rr})\,B\,C^2-A\,B^3))/(A\,B^3\,C^2) \tag{\%o41}$$

Kretschmann invariant

(%i42) `rinvariant();`

$$\frac{8(B\ (C_{rr})-(B_r)\ (C_r))^2}{B^6\,C^2}+\frac{4(C_r-B)^2\,(C_r+B)^2}{B^4\,C^4}+\frac{8(A_r)^2\,(C_r)^2}{A^2\,B^4\,C^2}+\frac{4((A_r)\ (B_r)-(A_{rr})\,B)^2}{A^2\,B^6} \tag{\%o42}$$

Einstein Tensor

(%i45)
```
ein:zeromatrix(dim,dim)$
einstein(false)$
for i thru dim do for j:i thru dim do
if ein[i,j]≠0 then
ishow('G([ξ[i]],[ξ[j]])=ein[i,j])$
```

$$G_t^t=-\frac{2BC\ (C_{rr})+B\,(C_r)^2-2\,(B_r)\,C\ (C_r)-B^3}{B^3\,C^2} \tag{\%t45}$$

$$G_r^r=-\frac{A\,(C_r)^2+2\,(A_r)\,C\ (C_r)-A\,B^2}{A\,B^2\,C^2} \tag{\%t45}$$

$$G_\theta^\theta=-\frac{AB\ (C_{rr})-A\ (B_r)\ (C_r)+(A_r)\,B\ (C_r)-(A_r)\ (B_r)\,C+(A_{rr})\,BC}{A\,B^3C} \tag{\%t45}$$

$$G_\phi^\phi=-\frac{AB\ (C_{rr})-A\ (B_r)\ (C_r)+(A_r)\,B\ (C_r)-(A_r)\ (B_r)\,C+(A_{rr})\,BC}{A\,B^3C} \tag{\%t45}$$

Einstein Tensor all down

(%i48)
```
lein:zeromatrix(dim,dim)$
leinstein(false)$
for i thru dim do for j:i thru dim do
if lein[i,j]≠0 then
ishow('G([ξ[i],ξ[j]],[])=lein[i,j])$
```

$$G_{tt}=-\frac{A^2\left(2BC\ (C_{rr})+B\,(C_r)^2-2\,(B_r)\,C\ (C_r)-B^3\right)}{B^3\,C^2} \tag{\%t48}$$

$$G_{rr}=\frac{A\,(C_r)^2+2\,(A_r)\,C\ (C_r)-A\,B^2}{A\,C^2} \tag{\%t48}$$

$$G_{\theta\theta}=\frac{C\ (AB\ (C_{rr})-A\ (B_r)\ (C_r)+(A_r)\,B\ (C_r)-(A_r)\ (B_r)\,C+(A_{rr})\,BC)}{A\,B^3} \tag{\%t48}$$

$$G_{\phi\phi}=\frac{C\ (AB\ (C_{rr})-A\ (B_r)\ (C_r)+(A_r)\,B\ (C_r)-(A_r)\ (B_r)\,C+(A_{rr})\,BC)\ \sin(\theta)^2}{A\,B^3} \tag{\%t48}$$

Returns a list of the unique differential equations

(%i49) `map(ldisp,efe:findde(ein,2))$`

$$2BC\ (C_{rr}) + B\,(C_r)^2 - 2\,(B_r)\,C\ (C_r) - B^3 \quad (\%t49)$$

$$A\,(C_r)^2 + 2\,(A_r)\,C\ (C_r) - A\,B^2 \quad (\%t50)$$

$$AB\ (C_{rr}) - A\ (B_r)\ (C_r) + (A_r)\,B\ (C_r) - (A_r)\ (B_r)\,C + (A_{rr})\,BC \quad (\%t51)$$

(%i52) `deindex;`

$$[[1,1],[2,2],[3,3]] \quad (\%o52)$$

Based on Wikipedia Article Spin connection

Verify $\mathrm{g}_{\mu\,\nu} = e_{\mu}{}^{a}\, e_{\nu}{}^{b}\, \eta_{a\,b}$

(%i53) `kill(labels,I,O,M)$`

(%i1) `imetric(g)$`

(%i3)
```
decsym(g,2,0,[sym(all)],[])$
decsym(g,0,2,[],[sym(all)])$
```

(%i4) `metricconvert:true$`

(%i5) `ishow(Eq:O([μ,ν])=I([μ],[a])*I([ν],[b])*M([a,b]))$`

$$O_{\mu\nu} = M_{ab}\, I_{\mu}^{a}\, I_{\nu}^{b} \quad (\%t5)$$

(%i6) `indices(Eq);`

$$[[\nu,\mu],[b,a]] \quad (\%o6)$$

(%i7) `Ver:ic_convert(Eq)$`

(%i10)
```
O:zeromatrix(dim,dim)$
I:fri$
M:lfg$
```

(%i11) `ev(Ver)$`

(%i12) `ldisplay(O)$`

$$O = \begin{pmatrix} A^2 & 0 & 0 & 0 \\ 0 & -B^2 & 0 & 0 \\ 0 & 0 & -C^2 & 0 \\ 0 & 0 & 0 & -C^2 \sin(\theta)^2 \end{pmatrix} \quad (\%t12)$$

(%i13) `is(O=lg);`

$$\text{true} \quad (\%o13)$$

Raise coordinate index

(%i14) `kill(I,O)$`

(%i15) `ishow(Eq:O([],[μ,a])=g([],[μ,ν])*I([ν],[a]))$`

$$O^{\mu a} = g^{\mu\nu}\, I_{\nu}^{a} \tag{\%t15}$$

(%i16) `indices(Eq);`

$$[[a, \mu], [\nu]] \tag{\%o16}$$

(%i17) `Raise:ic_convert(Eq)$`

(%i19) `O:zeromatrix(dim,dim)$`
`I:fri$`

(%i20) `ev(Raise)$`

(%i21) `ldisplay(O)$`

$$O = \begin{pmatrix} \frac{1}{A} & 0 & 0 & 0 \\ 0 & -\frac{1}{B} & 0 & 0 \\ 0 & 0 & -\frac{1}{C} & 0 \\ 0 & 0 & 0 & -\frac{1}{C\,\sin(\theta)} \end{pmatrix} \tag{\%t21}$$

Save for later

(%i22) `Raised:O$`

Lower tetrad index

(%i23) `kill(I,O,M)$`

(%i24) `ishow(Eq:O([ν,a],[])=M([a,b])*I([ν],[b]))$`

$$O_{\nu a} = M_{ab}\, I_{\nu}^{b} \tag{\%t24}$$

(%i25) `indices(Eq);`

$$[[a, \nu], [b]] \tag{\%o25}$$

(%i26) `Lower:ic_convert(Eq)$`

(%i29) `O:zeromatrix(dim,dim)$`
`I:fri$`
`M:lfg$`

(%i30) `ev(Lower)$`

(%i31) `ldisplay(O)$`

$$O = \begin{pmatrix} A & 0 & 0 & 0 \\ 0 & -B & 0 & 0 \\ 0 & 0 & -C & 0 \\ 0 & 0 & 0 & -C\,\sin(\theta) \end{pmatrix} \tag{\%t31}$$

Spin Connection 1

(%i32)
```
kill(I,J,ω)$
```

(%i33)
```
ishow(Eq:ω([μ],[a,b])=I([ν],[a])*ichr2([σ,μ],[ν])*
J([],[σ,b])+I([ν],[a])*idiff(J([],[ν,b]),μ))$
```

$$\omega_{\mu}^{ab} = \frac{g^{\nu\%1}\, J^{\sigma b}\, I_{\nu}^{a}\, \left(-g_{\sigma\mu,\%1} + g_{\sigma\%1,\mu} + g_{\mu\%1,\sigma}\right)}{2} + J_{,\mu}^{\nu b}\, I_{\nu}^{a} \quad (\%t33)$$

(%i34)
```
indices(Eq);
```

$$[[a, b, \mu], [\sigma, \%1, \nu]] \quad (\%o34)$$

(%i35)
```
SC1:ic_convert(Eq)$
```

(%i39)
```
kill(ω)$
array(ω,dim,dim,dim)$
I:fri$
J:Raised$
```

(%i40)
```
ev(SC1)$
```

(%i41)
```
cdisplay(ω)$
```

$$\omega_1 = \begin{pmatrix} 0 & -\frac{A_r}{B} & 0 & 0 \\ \frac{A_r}{B} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix} \omega_2 = \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix} \omega_3 = \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{C_r}{B} & 0 \\ 0 & -\frac{C_r}{B} & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix} \omega_4 = \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{(C_r)\,\sin(\theta)}{B} \\ 0 & 0 & 0 & \cos(\theta) \\ 0 & -\frac{(C_r)\,\sin(\theta)}{B} & -\cos(\theta) & 0 \end{pmatrix}$$

Spin Connection 2

(%i42)
```
kill(I,J,ω)$
```

(%i43)
```
ishow(Eq:ω([μ],[a,b])=I([ν],[a])*ichr2([σ,μ],[ν])*
J([],[σ,b])-J([],[ν,b])*idiff(I([ν],[a]),μ))$
```

$$\omega_{\mu}^{ab} = \frac{g^{\nu\%2}\, J^{\sigma b}\, I_{\nu}^{a}\, \left(-g_{\sigma\mu,\%2} + g_{\sigma\%2,\mu} + g_{\mu\%2,\sigma}\right)}{2} - J^{\nu b}\, I_{\nu,\mu}^{a} \quad (\%t43)$$

(%i44)
```
indices(Eq);
```

$$[[a, b, \mu], [\sigma, \%2, \nu]] \quad (\%o44)$$

(%i45)
```
SC2:ic_convert(Eq)$
```

(%i49)
```
kill(ω)$
array(ω,dim,dim,dim)$
I:fri$
J:Raised$
```

(%i50)
```
ev(SC2)$
```

(%i51) `cdisplay(ω)$`

$$\omega_1 = \begin{pmatrix} 0 & -\frac{A_r}{B} & 0 & 0 \\ \frac{A_r}{B} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix} \omega_2 = \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix} \omega_3 = \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{C_r}{B} & 0 \\ 0 & -\frac{C_r}{B} & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix} \omega_4 = \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{(C_r)\,\sin(\theta)}{B} \\ 0 & 0 & 0 & \cos(\theta) \\ 0 & -\frac{(C_r)\,\sin(\theta)}{B} & -\cos(\theta) & 0 \end{pmatrix}$$

Connection 1-form Θ

(%i52) `kill(Θ)$`

(%i53) `ishow(Eq:Θ([],[a,b])=ω([μ],[a,b])*dξ([],[μ]))$`

$$\Theta^{ab} = d\xi^{\mu}\,\omega_{\mu}^{ab} \tag{\%t53}$$

(%i54) `indices(Eq);`

$$[[b,a],[\mu]] \tag{\%o54}$$

(%i55) `SC3:ic_convert(Eq)$`

3 Use Cartan package

(%i2) `reset()$kill(allbut(ξ,dim,ω,Θ,SC3))$`

(%i1) `derivabbrev:true$`

(%i2) `ratprint:false$`

(%i3) `fpprintprec:5$`

(%i4) `if get('cartan,'version)=false then load(cartan)$`

(%i5) `if get('format,'version)=false then load(format)$`

(%i6) `depends([A,B,C],r)$`

(%i7) `assume(A>0,B>0,C>0)$`

Initialize Cartan package

(%i8) `init_cartan(ξ)$`

(%i9) `cartan_basis;`

$$[dt, dr, d\theta, d\phi] \tag{\%o9}$$

(%i10) `cartan_coords;`

$$[t, r, \theta, \phi] \tag{\%o10}$$

(%i11) `cartan_dim;`

$$4 \tag{\%o11}$$

(%i12) `extdim;`

$$4 \tag{\%o12}$$

Connection 1-form Θ

(%i14) `Θ:zeromatrix(dim,dim)$`
`dξ:cartan_basis$`

(%i15) `ev(SC3)$`

(%i16) `ldisplay(Θ)$`

$$\mathit{\Theta} = \begin{pmatrix} 0 & -\frac{(A_r)\,dt}{B} & 0 & 0 \\ \frac{(A_r)\,dt}{B} & 0 & \frac{(C_r)\,d\theta}{B} & \frac{(C_r)\,d\phi\,\sin(\theta)}{B} \\ 0 & -\frac{(C_r)\,d\theta}{B} & 0 & d\phi\,\cos(\theta) \\ 0 & -\frac{(C_r)\,d\phi\,\sin(\theta)}{B} & -d\phi\,\cos(\theta) & 0 \end{pmatrix} \tag{\%t16}$$

Exterior derivative of Connection 1-form $\mathrm{d}\Theta$

(%i17) `ldisplay(dΘ:fullratsimp(trigsimp(matrixmap(edit,ext_diff(Θ)))))$`

$$d\Theta=\begin{pmatrix} 0 & -\frac{((A_r)\,(B_r)-(A_{rr})B)\;dr\;dt}{B^2} & 0 & \\ \frac{((A_r)\,(B_r)-(A_{rr})B)\;dr\;dt}{B^2} & 0 & \frac{(B\,(C_{rr})-(B_r)\,(C_r))\;dr\;d\theta}{B^2} & d\phi\,((B\,(C_{rr})-(B_r)\,(C_r) \\ 0 & -\frac{(B\,(C_{rr})-(B_r)\,(C_r))\;dr\;d\theta}{B^2} & 0 & -d\theta \\ 0 & -\frac{d\phi\,((B\,(C_{rr})-(B_r)\,(C_r))\;dr\;\sin(\theta)+B\,(C_r)\;d\theta\;\cos(\theta))}{B^2} & d\theta\;d\phi\;\sin(\theta) & \end{pmatrix} \quad (\%t17)$$

Curvature 2-form Ω

(%i18) `matrix_element_mult:"~"$`

(%i19) `ldisplay(Ω:fullratsimp(matrixmap(edit,dΘ+Θ.Θ)))$`

$$\Omega=\begin{pmatrix} 0 & -\frac{((A_r)\,(B_r)-(A_{rr})B)\;dr\;dt}{B^2} & -\frac{(A_r)\,(C_r)\;dt\;d\theta}{B^2} & -\frac{(A_r)\,(C \\ \frac{((A_r)\,(B_r)-(A_{rr})B)\;dr\;dt}{B^2} & 0 & \frac{(B\,(C_{rr})-(B_r)\,(C_r))\;dr\;d\theta}{B^2} & d\phi\,((B\,(C_{rr})-(B_r)\,(C_r)) \\ \frac{(A_r)\,(C_r)\;dt\;d\theta}{B^2} & -\frac{(B\,(C_{rr})-(B_r)\,(C_r))\;dr\;d\theta}{B^2} & 0 & -\frac{((C_r)^2+ \\ \frac{(A_r)\,(C_r)\;dt\;d\phi\;\sin(\theta)}{B^2} & -\frac{d\phi\,((B\,(C_{rr})-(B_r)\,(C_r))\;dr\;\sin(\theta)+2B\,(C_r)\;d\theta\;\cos(\theta))}{B^2} & \frac{\left((C_r)^2+B^2\right)\;d\theta\;d\phi\;\sin(\theta)}{B^2} & \end{pmatrix} \quad (\%t19)$$

(%i20) `matrix_element_mult:"."$`

(%i21) `block([matrix_element_mult:"~"], Ω:fullratsimp(matrixmap(edit,dΘ+Θ.Θ)));`

$$\begin{pmatrix} 0 & -\frac{((A_r)\,(B_r)-(A_{rr})B)\;dr\;dt}{B^2} & -\frac{(A_r)\,(C_r)\;dt\;d\theta}{B^2} & -\frac{(A_r)\,(C_r)\;dt\;d}{B^2} \\ \frac{((A_r)\,(B_r)-(A_{rr})B)\;dr\;dt}{B^2} & 0 & \frac{(B\,(C_{rr})-(B_r)\,(C_r))\;dr\;d\theta}{B^2} & \frac{d\phi\,((B\,(C_{rr})-(B_r)\,(C_r))\;dr\;\sin}{B^2} \\ \frac{(A_r)\,(C_r)\;dt\;d\theta}{B^2} & -\frac{(B\,(C_{rr})-(B_r)\,(C_r))\;dr\;d\theta}{B^2} & 0 & -\frac{\left((C_r)^2+B^2\right)\;d}{B^2} \\ \frac{(A_r)\,(C_r)\;dt\;d\phi\;\sin(\theta)}{B^2} & -\frac{d\phi\,((B\,(C_{rr})-(B_r)\,(C_r))\;dr\;\sin(\theta)+2B\,(C_r)\;d\theta\;\cos(\theta))}{B^2} & \frac{\left((C_r)^2+B^2\right)\;d\theta\;d\phi\;\sin(\theta)}{B^2} & 0 \end{pmatrix} \quad (\%o21)$$

(%i22)
```
for i thru dim do for j:i thru dim do
if Ω[i,j]≠0 then ldisplay('Ω[ξ[i],ξ[j]]=fullratsimp(Ω[i,j]))$
```

$$\Omega_{t,r}=-\frac{((A_r)\;(B_r)-(A_{rr})\;B)\;\;dr\;dt}{B^2} \quad (\%t22)$$

$$\Omega_{t,\theta}=-\frac{(A_r)\;(C_r)\;\;dt\;d\theta}{B^2} \quad (\%t23)$$

$$\Omega_{t,\phi}=-\frac{(A_r)\;(C_r)\;\;dt\;d\phi\;\sin(\theta)}{B^2} \quad (\%t24)$$

$$\Omega_{r,\theta}=\frac{(B\;(C_{rr})-(B_r)\;(C_r))\;\;dr\;d\theta}{B^2} \quad (\%t25)$$

$$\Omega_{r,\phi}=\frac{d\phi\;((B\;(C_{rr})-(B_r)\;(C_r))\;\;dr\;\sin(\theta)+2B\;(C_r)\;\;d\theta\;\cos(\theta))}{B^2} \quad (\%t26)$$

$$\Omega_{\theta,\phi}=-\frac{\left((C_r)^2+B^2\right)\;d\theta\;d\phi\;\sin(\theta)}{B^2} \quad (\%t27)$$

Clean up

(%i31)
```
forget(0≤r)$
forget(0≤θ,θ≤π)$
forget(0≤sin(θ))$
forget(0≤ϕ,ϕ≤2*π)$
```

(%i32)
```
elapsed_real_time();
```

$$38.65 \tag{\%o32}$$

(%i33)
```
elapsed_run_time();
```

$$3.422 \tag{\%o33}$$