

```
% Gravitational Physics
% Lectures by Ruth Gregory
```

```
load_package excalc$
```

```
pform {A,B,C}=0$
fdomain A=A(r),B=B(r),C=C(r)$
```

```
write "Define the Ansatz Schwarzschild coframe"$
coframe o(t)      = A          * d t,
      o(r)      = B          * d r,
      o(theta)  = C          * d theta,
      o(phi)    = C * sin(theta) * d phi
with metric g = + o(t) * o(t) - o(r) * o(r)
               - o(theta) * o(theta) - o(phi) * o(phi)$
```

```
frame e$
displayframe;
DETM!*;
```

```
on fancy;
on nero$
factor o,^$
```

```
write "Verify"$
e(-k) _| o(l);
```

```
clear omega$
riemannconx omega$
write "Display the connection form"$
omega(k,-l);
```

```
write "Display the connection form in Matrix"$
coords := {t, r, theta, phi}$
matrix Momega(4, 4)$
for k := 1:4 do for l := 1:4 do
Momega(k, l) := omega(part(coords, k), part(coords, l))$
Momega;
```

```
clear curv,riemann,ricci,riccisc$
pform curv(k,l)=2,{riemann(a,b,c,d),ricci(a,b),riccisc}=0$
```

```
index_symmetries curv(k,l): antisymmetric,
      riemann(k,l,m,n): antisymmetric in {k,l},{m,n}
                        symmetric in {{k,l},{m,n}},
      ricci(k,l): symmetric;
```

```
write "Display the curvature form"$
curv(k,-l) := d omega(k,-l) + omega(k,-m) ^ omega(m,-l);
```

```
write "Display the curvature form in Matrix"$
linelength 200$
matrix Mcurv(4, 4)$
for k := 1:4 do for l := 1:4 do
Mcurv(k, l) := curv(part(coords, k), part(coords, l))$
Mcurv;
```

```
write "Display the Riemann Tensor all up"$
```

```

riemann(a,b,c,d) := e(d) _| ( e(c) _| curv(a,b) );
write "Display the Riemann Tensor"$
riemann(a,-b,-c,-d);
write "Display the Riemann Tensor all down"$
riemann(-a,-b,-c,-d);

write "Display the Ricci Tensor"$
ricci(-a,-b) := riemann(c,-a,-d,-b) * g(-c,d);

write "Display the Ricci Scalar"$
riccisc := ricci(-a,-b) * g(a,b);

write "Display the Einstein Tensor"$
clear einstein$
pform einstein(a)=3$
einstein(-a) := (1/2) * curv(b,c) ^ #( o(-b) ^ o(-c) ^ o(-a) );

showtime;
end;

*** .*. redefined

*** x redefined

*** ^ redefined

```

Define the Ansatz Schwarzschild coframe

$$\begin{aligned}
 o^t &= dt a \\
 o^r &= dr b \\
 o^\theta &= d\theta c \\
 o^\phi &= d\phi \sin(\theta) c
 \end{aligned}$$

-1

Verify

$$ns_t^t := 1$$

$$ns_r^r := 1$$

$$ns_\theta^\theta := 1$$

$$ns_\phi^\phi := 1$$

Display the connection form

$$\omega^r_t := \frac{o^t \frac{\partial a}{\partial r}}{a b}$$

$$\omega^t_r := \frac{o^t \frac{\partial a}{\partial r}}{a b}$$

$$\omega^\theta_r := \frac{o^\theta \frac{\partial c}{\partial r}}{b c}$$

$$\omega^\phi_r := \frac{o^\phi \frac{\partial c}{\partial r}}{b c}$$

$$\omega^r_\theta := \frac{-o^\theta \frac{\partial c}{\partial r}}{b c}$$

$$\omega^\phi_\theta := \frac{o^\phi \cos(\theta)}{\sin(\theta) c}$$

$$\omega^r_\phi := \frac{-o^\phi \frac{\partial c}{\partial r}}{b c}$$

$$\omega^\theta_\phi := \frac{-o^\phi \cos(\theta)}{\sin(\theta) c}$$

Display the connection form in Matrix

$$\begin{pmatrix} 0 & \frac{-o^t \frac{\partial a}{\partial r}}{a b} & 0 & 0 \\ \frac{o^t \frac{\partial a}{\partial r}}{a b} & 0 & \frac{o^\theta \frac{\partial c}{\partial r}}{b c} & \frac{o^\phi \frac{\partial c}{\partial r}}{b c} \\ 0 & \frac{-o^\theta \frac{\partial c}{\partial r}}{b c} & 0 & \frac{o^\phi \cos(\theta)}{\sin(\theta) c} \\ 0 & \frac{-o^\phi \frac{\partial c}{\partial r}}{b c} & \frac{-o^\phi \cos(\theta)}{\sin(\theta) c} & 0 \end{pmatrix}$$

Display the curvature form

$$\text{curv}_t^r := \frac{o^t \wedge o^r \left(-\frac{\partial^2 a}{\partial r^2} b + \frac{\partial a}{\partial r} \frac{\partial b}{\partial r} \right)}{a b^3}$$

$$\text{curv}_t^\theta := \frac{-o^t \wedge o^\theta \frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c}$$

$$\text{curv}_t^\phi := \frac{-o^t \wedge o^\phi \frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c}$$

$$\text{curv}_r^t := \frac{o^t \wedge o^r \left(-\frac{\partial^2 a}{\partial r^2} b + \frac{\partial a}{\partial r} \frac{\partial b}{\partial r} \right)}{a b^3}$$

$$\text{curv}_r^\theta := \frac{o^r \wedge o^\theta \left(-\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} + \frac{\partial^2 c}{\partial r^2} b \right)}{b^3 c}$$

$$\text{curv}_r^\phi := \frac{o^r \wedge o^\phi \left(-\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} + \frac{\partial^2 c}{\partial r^2} b \right)}{b^3 c}$$

$$\text{curv}_\theta^t := \frac{-o^t \wedge o^\theta \frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c}$$

$$\text{curv}_\theta^r := \frac{o^r \wedge o^\theta \left(\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} - \frac{\partial^2 c}{\partial r^2} b \right)}{b^3 c}$$

$$\text{curv}_\theta^\phi := \frac{o^\theta \wedge o^\phi \left(\left(\frac{\partial c}{\partial r} \right)^2 - b^2 \right)}{b^2 c^2}$$

$$\text{curv}_\phi^t := \frac{-o^t \wedge o^\phi \frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c}$$

$$\text{curv}_\phi^r := \frac{o^r \wedge o^\phi \left(\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} - \frac{\partial^2 c}{\partial r^2} b \right)}{b^3 c}$$

$$\text{curv}_\phi^\theta := \frac{o^\theta \wedge o^\phi \left(-\left(\frac{\partial c}{\partial r} \right)^2 + b^2 \right)}{b^2 c^2}$$

Display the curvature form in Matrix

$$\begin{pmatrix} 0 & \frac{o^t \wedge o^r \left(\frac{\partial^2 a}{\partial r^2} b - \frac{\partial a}{\partial r} \frac{\partial b}{\partial r} \right)}{a b^3} & \frac{o^t \wedge o^\theta \frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c} & \frac{o^t \wedge o^\phi \frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c} \\ \frac{o^t \wedge o^r \left(-\frac{\partial^2 a}{\partial r^2} b + \frac{\partial a}{\partial r} \frac{\partial b}{\partial r} \right)}{a b^3} & 0 & \frac{o^r \wedge o^\theta \left(-\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} + \frac{\partial^2 c}{\partial r^2} b \right)}{b^3 c} & \frac{o^r \wedge o^\phi \left(-\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} + \frac{\partial^2 c}{\partial r^2} b \right)}{b^3 c} \\ \frac{-o^t \wedge o^\theta \frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c} & \frac{o^r \wedge o^\theta \left(\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} - \frac{\partial^2 c}{\partial r^2} b \right)}{b^3 c} & 0 & \frac{o^\theta \wedge o^\phi \left(\left(\frac{\partial c}{\partial r} \right)^2 - b^2 \right)}{b^2 c^2} \\ \frac{-o^t \wedge o^\phi \frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c} & \frac{o^r \wedge o^\phi \left(\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} - \frac{\partial^2 c}{\partial r^2} b \right)}{b^3 c} & \frac{o^\theta \wedge o^\phi \left(-\left(\frac{\partial c}{\partial r} \right)^2 + b^2 \right)}{b^2 c^2} & 0 \end{pmatrix}$$

Display the Riemann Tensor all up

$$\text{riemann}^{t r t r} := \frac{-\frac{\partial^2 a}{\partial r^2} b + \frac{\partial a}{\partial r} \frac{\partial b}{\partial r}}{a b^3}$$

$$\text{riemann}^{t \theta t \theta} := \frac{-\frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c}$$

$$\text{riemann}^{r \theta r \theta} := \frac{-\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} + \frac{\partial^2 c}{\partial r^2} b}{b^3 c}$$

$$\text{riemann}^{t \phi t \phi} := \frac{-\frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c}$$

$$\text{riemann}^{r \phi r \phi} := \frac{-\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} + \frac{\partial^2 c}{\partial r^2} b}{b^3 c}$$

$$\text{riemann}^{\theta \phi \theta \phi} := \frac{\left(\frac{\partial c}{\partial r}\right)^2 - b^2}{b^2 c^2}$$

Display the Riemann Tensor

$$\text{riemann}^r_{t r t} := \frac{\frac{\partial^2 a}{\partial r^2} b - \frac{\partial a}{\partial r} \frac{\partial b}{\partial r}}{a b^3}$$

$$\text{riemann}^t_{r r t} := \frac{\frac{\partial^2 a}{\partial r^2} b - \frac{\partial a}{\partial r} \frac{\partial b}{\partial r}}{a b^3}$$

$$\text{riemann}^\theta_{t \theta t} := \frac{\frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c}$$

$$\text{riemann}^t_{\theta \theta t} := \frac{\frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c}$$

$$\text{riemann}^\phi_{t \phi t} := \frac{\frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c}$$

$$\text{riemann}^t_{\phi \phi t} := \frac{\frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c}$$

$$\text{riemann}^r_{t t r} := \frac{-\frac{\partial^2 a}{\partial r^2} b + \frac{\partial a}{\partial r} \frac{\partial b}{\partial r}}{a b^3}$$

$$\text{riemann}^t_{r t r} := \frac{-\frac{\partial^2 a}{\partial r^2} b + \frac{\partial a}{\partial r} \frac{\partial b}{\partial r}}{a b^3}$$

$$\text{riemann}^\theta_{r \theta r} := \frac{\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} - \frac{\partial^2 c}{\partial r^2} b}{b^3 c}$$

$$\begin{aligned} \text{riemann}^r_{\theta\theta r} &:= \frac{-\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} + \frac{\partial^2 c}{\partial r^2} b}{b^3 c} \\ \text{riemann}^\phi_{r\phi r} &:= \frac{\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} - \frac{\partial^2 c}{\partial r^2} b}{b^3 c} \\ \text{riemann}^r_{\phi\phi r} &:= \frac{-\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} + \frac{\partial^2 c}{\partial r^2} b}{b^3 c} \\ \text{riemann}^\theta_{tt\theta} &:= \frac{-\frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c} \\ \text{riemann}^t_{\theta t\theta} &:= \frac{-\frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c} \\ \text{riemann}^\theta_{rr\theta} &:= \frac{-\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} + \frac{\partial^2 c}{\partial r^2} b}{b^3 c} \\ \text{riemann}^r_{\theta r\theta} &:= \frac{\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} - \frac{\partial^2 c}{\partial r^2} b}{b^3 c} \\ \text{riemann}^\phi_{\theta\phi\theta} &:= \frac{-\left(\frac{\partial c}{\partial r}\right)^2 + b^2}{b^2 c^2} \\ \text{riemann}^\theta_{\phi\phi\theta} &:= \frac{\left(\frac{\partial c}{\partial r}\right)^2 - b^2}{b^2 c^2} \\ \text{riemann}^\phi_{tt\phi} &:= \frac{-\frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c} \\ \text{riemann}^t_{\phi t\phi} &:= \frac{-\frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c} \\ \text{riemann}^\phi_{rr\phi} &:= \frac{-\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} + \frac{\partial^2 c}{\partial r^2} b}{b^3 c} \\ \text{riemann}^r_{\phi r\phi} &:= \frac{\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} - \frac{\partial^2 c}{\partial r^2} b}{b^3 c} \\ \text{riemann}^\phi_{\theta\theta\phi} &:= \frac{\left(\frac{\partial c}{\partial r}\right)^2 - b^2}{b^2 c^2} \\ \text{riemann}^\theta_{\phi\theta\phi} &:= \frac{-\left(\frac{\partial c}{\partial r}\right)^2 + b^2}{b^2 c^2} \end{aligned}$$

Display the Riemann Tensor all down

$$\begin{aligned} \text{riemann}_{t r t r} &:= \frac{-\frac{\partial^2 a}{\partial r^2} b + \frac{\partial a}{\partial r} \frac{\partial b}{\partial r}}{a b^3} \\ \text{riemann}_{t \theta t \theta} &:= \frac{-\frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c} \end{aligned}$$

$$\text{riemann}_{r\ \theta\ r\ \theta} := \frac{-\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} + \frac{\partial^2 c}{\partial r^2} b}{b^3 c}$$

$$\text{riemann}_{t\ \phi\ t\ \phi} := \frac{-\frac{\partial a}{\partial r} \frac{\partial c}{\partial r}}{a b^2 c}$$

$$\text{riemann}_{r\ \phi\ r\ \phi} := \frac{-\frac{\partial b}{\partial r} \frac{\partial c}{\partial r} + \frac{\partial^2 c}{\partial r^2} b}{b^3 c}$$

$$\text{riemann}_{\theta\ \phi\ \theta\ \phi} := \frac{\left(\frac{\partial c}{\partial r}\right)^2 - b^2}{b^2 c^2}$$

Display the Ricci Tensor

$$\text{ricci}_{t\ t} := \frac{\frac{\partial^2 a}{\partial r^2} b c - \frac{\partial a}{\partial r} \frac{\partial b}{\partial r} c + 2 \frac{\partial a}{\partial r} \frac{\partial c}{\partial r} b}{a b^3 c}$$

$$\text{ricci}_{r\ r} := \frac{-\frac{\partial^2 a}{\partial r^2} b c + \frac{\partial a}{\partial r} \frac{\partial b}{\partial r} c + 2 \frac{\partial b}{\partial r} \frac{\partial c}{\partial r} a - 2 \frac{\partial^2 c}{\partial r^2} a b}{a b^3 c}$$

$$\text{ricci}_{\theta\ \theta} := \frac{-\frac{\partial a}{\partial r} \frac{\partial c}{\partial r} b c + \frac{\partial b}{\partial r} \frac{\partial c}{\partial r} a c - \frac{\partial^2 c}{\partial r^2} a b c - \left(\frac{\partial c}{\partial r}\right)^2 a b + a b^3}{a b^3 c^2}$$

$$\text{ricci}_{\phi\ \phi} := \frac{-\frac{\partial a}{\partial r} \frac{\partial c}{\partial r} b c + \frac{\partial b}{\partial r} \frac{\partial c}{\partial r} a c - \frac{\partial^2 c}{\partial r^2} a b c - \left(\frac{\partial c}{\partial r}\right)^2 a b + a b^3}{a b^3 c^2}$$

Display the Ricci Scalar

$$\text{riccisc} := \frac{2 \left(\frac{\partial^2 a}{\partial r^2} b c^2 - \frac{\partial a}{\partial r} \frac{\partial b}{\partial r} c^2 + 2 \frac{\partial a}{\partial r} \frac{\partial c}{\partial r} b c - 2 \frac{\partial b}{\partial r} \frac{\partial c}{\partial r} a c + 2 \frac{\partial^2 c}{\partial r^2} a b c + \left(\frac{\partial c}{\partial r}\right)^2 a b - a b^3 \right)}{a b^3 c^2}$$

Display the Einstein Tensor

$$\text{einstein}_t := \frac{o^r \wedge o^\theta \wedge o^\phi \left(-2 \frac{\partial b}{\partial r} \frac{\partial c}{\partial r} c + 2 \frac{\partial^2 c}{\partial r^2} b c + \left(\frac{\partial c}{\partial r}\right)^2 b - b^3 \right)}{b^3 c^2}$$

$$\text{einstein}_r := \frac{o^t \wedge o^\theta \wedge o^\phi \left(-2 \frac{\partial a}{\partial r} \frac{\partial c}{\partial r} c - \left(\frac{\partial c}{\partial r}\right)^2 a + a b^2 \right)}{a b^2 c^2}$$

$$\text{einstein}_\theta := \frac{o^t \wedge o^r \wedge o^\phi \left(\frac{\partial^2 a}{\partial r^2} b c - \frac{\partial a}{\partial r} \frac{\partial b}{\partial r} c + \frac{\partial a}{\partial r} \frac{\partial c}{\partial r} b - \frac{\partial b}{\partial r} \frac{\partial c}{\partial r} a + \frac{\partial^2 c}{\partial r^2} a b \right)}{a b^3 c}$$

$$\text{einstein}_\phi:=\frac{o^t\wedge o^r\wedge o^\theta\left(-\frac{\partial^2a}{\partial r^2}bc+\frac{\partial a}{\partial r}\frac{\partial b}{\partial r}c-\frac{\partial a}{\partial r}\frac{\partial c}{\partial r}b+\frac{\partial b}{\partial r}\frac{\partial c}{\partial r}a-\frac{\partial^2c}{\partial r^2}ab\right)}{ab^3c}$$

Time: 24921 ms plus GC time: 4 ms