ONE FLUX EXAMPLE TWO WAYS

Based on Dr. Bevin Maultsby Playlist One flux example two ways: using Stokes' and the Divergence Theorem

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2) `info:build_info()$info@version;`

(%o2)

5.38.1

```
(%i2) reset()$kill(all)$
(%i1) derivabbrev:true$
(%i2) ratprint:false$
(%i3) fpprintprec:5$
(%i4) load(linearalgebra)$
(%i5) if get('draw,'version)=false then load(draw)$
(%i6) wxplot_size:[1024,768]$
(%i7) set_draw_defaults(xtics=1,ytics=1,ztics=1,xyplane=0,nticks=100,
      xaxis=true,xaxis_type=dots,xaxis_width=3,
      yaxis=true,yaxis_type=dots,yaxis_width=3,
      zaxis=true,zaxis_type=dots,zaxis_width=3,
      background_color=light_gray)$
(%i8) if get('vect,'version)=false then load(vect)$
(%i9) norm(u):=block(ratsimp(radcan(√(u.u))))$
(%i10) normalize(v):=block(v/norm(v))$
(%i11) angle(u,v):=block([junk:radcan(√((u.u)*(v.v)))],acos(u.v/junk))$
(%i12) mycross(va,vb):=[va[2]*vb[3]-va[3]*vb[2],va[3]*vb[1]-va[1]*vb[3],va[1]*vb[2]-va[2]*vb[1]]$
(%i13) if get('cartan,'version)=false then load(cartan)$
(%i14) declare(trigsimp,evfun)$
```

Let S be the surface $z = x^2 + y^2$, $0 \leq z \leq 16$, oriented with outward-pointing normal vectors.

Let $\vec{F}(x, y, z) = \langle 2\,y, -2\,x, -8\,x^2 + 12\,y + cos(z^2)\rangle$.

Compute $\iint_S \nabla \times \vec{F} \cdot \mathrm{d}\vec{S}$ using ...

(a) Stokes' theorem

(b) the Divergence theorem

Stokes' theorem

$$\iint_S \nabla \times \vec{F} \cdot \mathrm{d}\vec{S} = \oint_{\partial S} \vec{F} \cdot \mathrm{d}\vec{r}$$

Define the space $\mathbb{R}^3$

```
(%i15) ζ:[x,y,z]$
(%i16) dim:length(ζ)$
(%i17) scalefactors(ζ)$
(%i18) init_cartan(ζ)$
```

Vector field $\vec{F} \in \mathbb{R}^3$

```
(%i19) ldisplay(F:[2*y,-2*x,-8*x^2+12*y+cos(z^2)])$
```

$$F = [2y, -2x, \cos\left(z^2\right) + 12y - 8x^2] \quad (\%t19)$$

3D Direction field

```
(%i21) /* vector origins are (x,y,z)| x,y=1,...,5 */
       coord:setify(makelist(k,k,-4,4))$
       points3d:listify(cartesian_product(coord,coord,coord))$
(%i23) /* compute vectors at the given points */
       define(vf3d(x,y,z),vector(ζ,F))$
       vect3:makelist(vf3d(k[1],k[2],k[3]),k,points3d)$
```

(%i24) `wxdraw3d([head_length=0.1,color=blue,head_angle=25,unit_vectors=true],vect3)$`



(%t24)

Calculate $\nabla \times \vec{F} \in \mathbb{R}^3$

(%i25) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = [12, 16x, -4] \tag{\%t25}$$

Work form $\alpha = F^\flat \in \mathcal{A}^1(\mathbb{R}^3)$

(%i26) `ldisplay(α:F.cartan_basis)$`

$$\alpha = \left(\cos\left(z^2\right) + 12y - 8x^2\right)\ dz - 2x\ dy + 2y\ dx \tag{\%t26}$$

Calculate $\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

(%i27) `ldisplay(dα:edit(ext_diff(α)))$`

$$d\alpha = 12 dy\ dz - 16x\ dx\ dz - 4 dx\ dy \tag{\%t27}$$

Calculate $\nabla \cdot \vec{F} \in \mathbb{R}$

(%i28) `ldisplay(divF:ev(express(div(F)),diff))$`

$$divF = -2z\ \sin\left(z^2\right) \tag{\%t28}$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i29)
```
ldisplay(β:F[1]*cartan_basis[2]~cartan_basis[3]+
F[2]*cartan_basis[3]~cartan_basis[1]+
F[3]*cartan_basis[1]~cartan_basis[2])$
```

$$\beta = 2y\ dy\ dz + 2x\ dx\ dz + \left(\cos\left(z^2\right) + 12y - 8x^2\right)\ dx\ dy \tag{\%t29}$$

Calculate $\mathrm{d}\beta \in \mathcal{A}^3(\mathbb{R}^3)$

(%i30) `ldisplay(dβ:edit(ext_diff(β)))$`

$$d\beta = -2z\ \sin\left(z^2\right)\ dx\ dy\ dz \tag{\%t30}$$

Draw the paraboloid and its boundary

(%i31)
```
wxdraw3d(axis_3d=false,xyplane=0,nticks=100,
x_voxel=20,y_voxel=20,z_voxel=20,
implicit(z=x^2+y^2,x,-4,4,y,-4,4,z,0,16),
line_width=3,color=black,
parametric(4*cos(-t),4*sin(-t),16,t,0,2*π))$
```



(%t31)

Curve $\vec{r} \in \mathbb{R}^3$

(%i32) `ldisplay(r:[4*cos(-t),4*sin(-t),16])$`

$$r = [4\cos(t), -4\sin(t), 16] \tag{\%t32}$$

Derivative of the curve $\vec{r}$

(%i33) `ldisplay(r\':diff(r,t))$`

$$r' = [-4\sin(t), -4\cos(t), 0] \tag{\%t33}$$

Calculate $\vec{F} \circ \vec{r}$

(%i34) `ldisplay(For:subst(map("=",ζ,r),F))$`

$$For = [-8\sin(t), -8\cos(t), -48\sin(t) - 128\cos(t)^2 + \cos(256)] \tag{\%t34}$$

Calculate $\vec{r}^{*}\alpha \in \mathcal{A}^1(\mathbb{R}^3)$

(%i35) `integrand:trigsimp(r\'|subst(map("=",ζ,r),α));`

$$32 \tag{integrand}$$

Calculate $\vec{F} \cdot \vec{r}' \in \mathbb{R}$

(%i36) `integrand:trigsimp(For.r\');`

$$32 \tag{integrand}$$

Flux integral

(%i37) `integrate(integrand,t,0,2*π);`

$$64\pi \tag{\%o37}$$

the Divergence theorem

$$\iint_S \nabla\times\vec{F}\cdot \mathrm{d}\vec{S}+\iint_L \nabla\times\vec{F}\cdot \mathrm{d}\vec{S}=\iiint_E \nabla\cdot(\nabla\times\vec{F})\mathrm{d}V=0$$

Work form $\gamma=(\nabla\times F)^\flat\in\mathcal{A}^1(\mathbb{R}^3)$

(%i38) `ldisplay(γ:curlF.cartan_basis)$`

$$\gamma=-4dz+16x\,dy+12dx \tag{\%t38}$$

Surface $\vec{r}\in\mathbb{R}^3$

(%i39) `ldisplay(r:[u*cos(v),u*sin(v),16])$`

$$r=[u\,\cos(v),u\,\sin(v),16] \tag{\%t39}$$

(%i40) `ldisplay(r_u:diff(r,u))$`

$$r_u=[\cos(v),\sin(v),0] \tag{\%t40}$$

(%i41) `ldisplay(r_v:diff(r,v))$`

$$r_v=[-u\,\sin(v),u\,\cos(v),0] \tag{\%t41}$$

Normal

(%i42) `ldisplay(n:trigsimp(mycross(r_u,r_v)))$`

$$n=[0,0,u] \tag{\%t42}$$

Integrand

(%i43) `integrand:n|γ;`

$$-4u \tag{integrand}$$

(%i44) `integrand:curlF.n;`

$$-4u \tag{integrand}$$

Flux integral

(%i45) `I:-'integrate('integrate(integrand,u,0,4),v,0,2*π)$`

(%i46) `ldisplay(I=box(ev(I,integrate)))$`

$$8\pi\int_0^4 u du=(64\pi) \tag{\%t46}$$