Explicit calculation of spin connection

Based on StackExchange Article Explicit calculation of spin connection through Cartan's first structure equation

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2) `info:build_info()$info@version;`

(%o2)

5.38.1

(%i2) `reset()$kill(all)$`

(%i1) `derivabbrev:true$`

(%i2) `ratprint:false$`

(%i3) `fpprintprec:5$`

(%i4) `if get('vect,'version)=false then load(vect)$`

(%i5) `if get('cartan,'version)=false then load(cartan)$`

(%i6) `if get('format,'version)=false then load(format)$`

(%i7) `norm(u):=block(ratsimp(radcan(√(u.u))))$`

(%i8) `normalize(v):=block(v/norm(v))$`

(%i9) `angle(u,v):=block([junk:radcan(√((u.u)*(v.v)))],acos(u.v/junk))$`

(%i10) `mycross(va,vb):=[va[2]*vb[3]-va[3]*vb[2],va[3]*vb[1]-va[1]*vb[3],va[1]*vb[2]-va[2]*vb[1]]$`

(%i11) `declare(trigsimp,evfun)$`

1 Jesse wrote:

Given the metric

$$\mathrm{d}s^2 = F(r)^2\,\mathrm{d}r^2 + r^2\,\mathrm{d}\theta^2 + r^2\sin^2(\theta)\,\mathrm{d}\phi^2,$$

I'm trying to find the corresponding spin connections $\omega^a{}_b$ using the first structure equation:

$$\mathrm{d}e + \omega e = 0$$

I found the vielbeins e and their exterior derivatives $\mathrm{d}e$:

$$\mathrm{d}e^1 = 0, \quad \mathrm{d}e^2 = \mathrm{d}r\mathrm{d}\theta, \quad \mathrm{d}e^3 = \sin(\theta)\mathrm{d}r\mathrm{d}\phi + r\cos(\theta)\mathrm{d}\theta\mathrm{d}\phi,$$

but I am stuck on actually working out the ω. I read through Zee's 'GR in a nutshell', and he does the same calculation but just says: "In general, write

$$\omega^a_b = \omega^a_{bc}e^c = \omega^a_{b\mu}dx^\mu$$

Plug this into the first structure equation and match terms. How do I actually go about calculating ω^1_2, ω^1_3, and ω^2_3 at this point?

```
(%i15) assume(0≤r)$
       assume(0≤θ,θ≤π)$
       assume(0≤sin(θ))$
       assume(0≤φ,φ≤2*π)$
(%i16) ξ:[r,θ,φ]$
(%i17) dim:length(ξ)$
```

Line Element

```
(%i18) depends(F,r)$
(%i19) assume(F>0)$
(%i20) ldisplay(ds^2=line_element:F^2*del(r)^2+r^2*del(θ)^2+r^2*sin(θ)^2*del(φ)^2)$
```

$$ds^2 = r^2\sin(\theta)^2\operatorname{del}(\phi)^2 + r^2\operatorname{del}(\theta)^2 + F^2\operatorname{del}(r)^2 \qquad (\%t20)$$

Covariant Metric Tensor

```
(%i24) lg:zeromatrix(dim,dim)$
       for i thru dim do
       lg[i,i]:factor(coeff(expand(line_element),del(ξ[i])^2))$
       for j thru dim do for k thru dim do
       if j≠k then lg[j,k]:factor(expand(ratsimp(1/2*coeff(coeff(expand(line_element),del(ξ[j])),del(ξ[k]
       ldisplay(lg)$
```

$$lg = \begin{pmatrix} F^2 & 0 & 0 \\ 0 & r^2 & 0 \\ 0 & 0 & r^2\sin(\theta)^2 \end{pmatrix} \qquad (\%t24)$$

Contravariant Metric Tensor

(%i25) `ldisplay(ug:invert(lg))$`

$$ug = \begin{pmatrix} \frac{1}{F^2} & 0 & 0 \\ 0 & \frac{1}{r^2} & 0 \\ 0 & 0 & \frac{1}{r^2 \sin(\theta)^2} \end{pmatrix} \tag{\%t25}$$

Line element

(%i26) `ldisplay(ds`^{2}`=diff(`ξ`).lg.transpose(diff(`ξ`)))$`

$$ds^2 = r^2 \sin(\theta)^2 \operatorname{del}(\phi)^2 + r^2 \operatorname{del}(\theta)^2 + F^2 \operatorname{del}(r)^2 \tag{\%t26}$$

Define the frame e

(%i29) `e[r]:`$\sqrt{}$`(ug)[1]$`
`e[`θ`]:`$\sqrt{}$`(ug)[2]$`
`e[`ϕ`]:`$\sqrt{}$`(ug)[3]$`

(%i30) `ldisplay(e:apply('matrix,[e[r],e[`θ`],e[`ϕ`]]))$`

$$e = \begin{pmatrix} \frac{1}{F} & 0 & 0 \\ 0 & \frac{1}{r} & 0 \\ 0 & 0 & \frac{1}{r \sin(\theta)} \end{pmatrix} \tag{\%t30}$$

Initialize cartan package

(%i31) `init_cartan(`ξ`)$`

(%i32) `cartan_basis;`

$$[dr, d\theta, d\phi] \tag{\%o32}$$

(%i33) `cartan_coords;`

$$[r, \theta, \phi] \tag{\%o33}$$

(%i34) `cartan_dim;`

$$3 \tag{\%o34}$$

(%i35) `extdim;`

$$3 \tag{\%o35}$$

Define the coframe ω

(%i40)
```
kill(ω)$
ω[r]:list_matrix_entries(√(lg).cartan_basis)[1]$
ω[θ]:list_matrix_entries(√(lg).cartan_basis)[2]$
ω[ϕ]:list_matrix_entries(√(lg).cartan_basis)[3]$
ldisplay(ω:[ω[r],ω[θ],ω[ϕ]])$
```

$$\omega = [F\;dr, r\;d\theta, r\;d\phi\;\sin(\theta)] \tag{\%t40}$$

Verify $\langle \underline{\omega}^{a} \mid \underline{e}_{b} \rangle = \delta^{a}{}_{b}$

(%i41)
```
genmatrix(lambda([i,j],e[ξ[i]]|ω[ξ[j]]),cartan_dim,cartan_dim);
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%o41}$$

Calculate the external derivative of the coframe $\mathrm{d}\omega$

(%i42)
```
ldisplay(dω:ext_diff(ω))$
```

$$d\omega = [0, dr\;d\theta, dr\;d\phi\;\sin(\theta) + r\;d\theta\;d\phi\;\cos(\theta)] \tag{\%t42}$$

2 JamalS wrote:

As you have, the first step is to identify:

$$e^r = F(r)\,\mathrm{d}r, e^\theta = r\,\mathrm{d}\theta, e^\phi = r\sin\theta\,\mathrm{d}\phi$$

The trick is to then take the derivatives but re-express them in terms of e again. Thus,

$$\mathrm{d}e^r = 0, \quad \mathrm{d}e^\theta = -\mathrm{d}\theta\wedge\mathrm{d}r = -\frac{1}{rF(r)}e^\theta\wedge e^r$$

and,

$$\mathrm{d}e^\phi = -\sin\theta\mathrm{d}\phi\wedge\mathrm{d}r - r\cos\theta\mathrm{d}\phi\wedge\mathrm{d}\theta = -\frac{1}{rF(r)}e^\phi\wedge e^r - \frac{\cot\theta}{r^2}e^\phi\wedge e^\theta.$$

Now let's take an example of using Cartan's first equation. We have $\mathrm{d}e^a + \omega^a_b\wedge e^b = 0$ and if we choose $a=\theta$ the equations read,

$$\frac{1}{rF(r)}e^\theta\wedge e^r = \omega^\theta_r\wedge e^r + \omega^\theta_\theta\wedge e^\theta + \omega^\theta_\phi\wedge e^\phi.$$

We have $\omega^\theta_\theta = 0$ by anti-symmetry. We can identify now $\omega^\theta_r = -\omega^r_\theta = \frac{1}{rF(r)}e^\theta$. Notice the last term we could choose $\omega^\theta_\phi = 0$ however Cartan's equations are a system of equations, so we are not free to make this choice yet without considering the other equations. We can at best say ω^θ_ϕ is proportional to $\mathrm{d}\phi$ to ensure $\omega^\theta_\phi\wedge e^\phi = 0$. As it turns out, we don't have $\omega^\theta_\phi = 0$ because of the $a=\phi$ equation, which will give you $\omega^\theta_\phi = -r^{-2}\cot\theta\,e^\phi$.

I hope this elucidates how to use Cartan's structure equation. Computing the Ricci tensor is then much simpler, as rather than solving for components you're just plugging in and computing.

Generic Connection 1-form Θ

```
(%i45) A:[a_1,a_2,a_3]$
       B:[b_1,b_2,b_3]$
       C:[c_1,c_2,c_3]$
(%i51) kill(Θ)$
       Θ:zeromatrix(dim,dim)$
       Θ[1,2]:-Θ[2,1]:A.cartan_basis$
       Θ[1,3]:-Θ[3,1]:B.cartan_basis$
       Θ[2,3]:-Θ[3,2]:C.cartan_basis$
       ldisplay(Θ)$
```

$$\Theta = \begin{pmatrix} 0 & -a_3\,d\phi - a_2\,d\theta - a_1\,dr & -b_3\,d\phi - b_2\,d\theta - b_1\,dr \\ a_3\,d\phi + a_2\,d\theta + a_1\,dr & 0 & -c_3\,d\phi - c_2\,d\theta - c_1\,dr \\ b_3\,d\phi + b_2\,d\theta + b_1\,dr & c_3\,d\phi + c_2\,d\theta + c_1\,dr & 0 \end{pmatrix} \qquad (\%t51)$$

Change matrix multiplication operator

```
(%i52) matrix_element_mult:"~"$
```

(%i53) `λ:list_matrix_entries(expand(Θ.ω))$`

(%i54) `map(ldisp,λ)$`

$$-b_2 r\, d\theta\, d\phi\, \sin(\theta) - b_1 r\, dr\, d\phi\, \sin(\theta) + a_3 r\, d\theta\, d\phi - a_1 r\, dr\, d\theta \tag{\%t54}$$

$$-c_2 r\, d\theta\, d\phi\, \sin(\theta) - c_1 r\, dr\, d\phi\, \sin(\theta) - F\, a_3 dr\, d\phi - F\, a_2 dr\, d\theta \tag{\%t55}$$

$$-c_3 r\, d\theta\, d\phi - F\, b_3 dr\, d\phi + c_1 r\, dr\, d\theta - F\, b_2 dr\, d\theta \tag{\%t56}$$

Restore matrix multiplication operator

(%i57) `matrix_element_mult:"*"$`

Cartan's First structural equation $\mathrm{d}\omega^{i} = \Theta_{j}{}^{i} \wedge \omega^{j}$

(%i58) `Eq:zeromatrix(dim,dim)$`

(%i59)
```
for i thru dim do for j thru dim do
Eq[i,j]:format(coeff(coeff(dω,cartan_basis[i]),cartan_basis[j])=
coeff(coeff(-λ,cartan_basis[i]),cartan_basis[j]),%list)$
```

(%i60) `Eqs:apply('append,list_matrix_entries(Eq))$`

(%i61) `linsol:linsolve(Eqs,append(A,B,C));`

solve: dependent equations eliminated: (1 27 26 25 2 3 13 14 15 10 17 19 16 11 12 20 21 18)

$$\left[a_1 = 0, a_2 = \frac{1}{F}, a_3 = 0, b_1 = 0, b_2 = 0, b_3 = \frac{\sin(\theta)}{F}, c_1 = 0, c_2 = 0, c_3 = \cos(\theta)\right] \tag{linsol}$$

(%i62) `ldisplay(λ:at(λ,linsol))$`

$$\lambda = [0, -dr\, d\theta, -dr\, d\phi\, \sin(\theta) - r\, d\theta\, d\phi\, \cos(\theta)] \tag{\%t62}$$

(%i63) `is(dω=-λ);`

$$\text{true} \tag{\%o63}$$

Update Connection 1-form Θ

(%i64) `ldisplay(Θ:at(Θ,linsol))$`

$$\Theta = \begin{pmatrix} 0 & -\frac{d\theta}{F} & -\frac{d\phi\, \sin(\theta)}{F} \\ \frac{d\theta}{F} & 0 & -d\phi\, \cos(\theta) \\ \frac{d\phi\, \sin(\theta)}{F} & d\phi\, \cos(\theta) & 0 \end{pmatrix} \tag{\%t64}$$

Update Connection 2-form $\mathrm{d}\Theta$

(%i65) `ldisplay(dΘ:expand(matrixmap(edit,ext_diff(Θ))))$`

$$d\Theta = \begin{pmatrix} 0 & \frac{(F_r)\, dr\, d\theta}{F^2} & \frac{(F_r)\, dr\, d\phi\, \sin(\theta)}{F^2} - \frac{d\theta\, d\phi\, \cos(\theta)}{F} \\ -\frac{(F_r)\, dr\, d\theta}{F^2} & 0 & d\theta\, d\phi\, \sin(\theta) \\ \frac{d\theta\, d\phi\, \cos(\theta)}{F} - \frac{(F_r)\, dr\, d\phi\, \sin(\theta)}{F^2} & -d\theta\, d\phi\, \sin(\theta) & 0 \end{pmatrix} \tag{\%t65}$$

Update coefficients

(%i68)
```
ldisplay(A:at(A,linsol))$
ldisplay(B:at(B,linsol))$
ldisplay(C:at(C,linsol))$
```

$$A=\left[0,\frac{1}{F},0\right] \tag{\%t66}$$

$$B=\left[0,0,\frac{\sin(\theta)}{F}\right] \tag{\%t67}$$

$$C=[0,0,\cos(\theta)] \tag{\%t68}$$

Change matrix multiplication operator

(%i69)
```
matrix_element_mult:"~"$
```

Cartan's Second structural equation: ${\Omega_{j}}^{i}=\mathrm{d}{\Theta_{j}}^{i}+{\Theta_{k}}^{i}\wedge{\Theta_{j}}^{k}$

Curvature 2-form Ω

(%i70)
```
ldisplay(Ω:matrixmap(edit,dΘ+Θ.Θ))$
```

$$\Omega=\begin{pmatrix} 0 & \frac{(F_r)\,dr\,d\theta}{F^2} & \frac{(F_r)\,dr\,d\phi\,\sin(\theta)}{F^2} \\ -\frac{(F_r)\,dr\,d\theta}{F^2} & 0 & d\theta\,d\phi\left(\sin(\theta)-\frac{\sin(\theta)}{F^2}\right) \\ -\frac{(F_r)\,dr\,d\phi\,\sin(\theta)}{F^2} & d\theta\,d\phi\left(\frac{\sin(\theta)}{F^2}-\sin(\theta)\right) & 0 \end{pmatrix} \tag{\%t70}$$

Restore matrix multiplication operator

(%i71)
```
matrix_element_mult:"*"$
```

Forms in terms of the coframe σ

(%i72)
```
Eqs:makelist(σ[ξ[i]]=ω[ξ[i]],i,1,cartan_dim);
```

$$[\sigma_r=F\,dr,\sigma_\theta=r\,d\theta,\sigma_\phi=r\,d\phi\,\sin(\theta)] \tag{Eqs}$$

(%i73)
```
linsol:linsolve(Eqs,cartan_basis);
```

$$\left[dr=\frac{\sigma_r}{F},d\theta=\frac{\sigma_\theta}{r},d\phi=\frac{\sigma_\phi}{r\,\sin(\theta)}\right] \tag{linsol}$$

Connection 1-form Θ

(%i74)
```
ldisplay(Θ:ev(Θ,linsol,fullratsimp))$
```

$$\Theta=\begin{pmatrix} 0 & -\frac{\sigma_\theta}{F\,r} & -\frac{\sigma_\phi}{F\,r} \\ \frac{\sigma_\theta}{F\,r} & 0 & -\frac{\cos(\theta)\,\sigma_\phi}{r\,\sin(\theta)} \\ \frac{\sigma_\phi}{F\,r} & \frac{\cos(\theta)\,\sigma_\phi}{r\,\sin(\theta)} & 0 \end{pmatrix} \tag{\%t74}$$

Curvature 2-form Ω

(%i75) ldisplay(Ω:ev(Ω,linsol,fullratsimp))$

$$\Omega = \begin{pmatrix} 0 & \frac{(F_r)\,\sigma_r\,\sigma_\theta}{F^3 r} & \frac{(F_r)\,\sigma_r\,\sigma_\phi}{F^3 r} \\ -\frac{(F_r)\,\sigma_r\,\sigma_\theta}{F^3 r} & 0 & \frac{\left(F^2-1\right)\sigma_\theta\,\sigma_\phi}{F^2\,r^2} \\ -\frac{(F_r)\,\sigma_r\,\sigma_\phi}{F^3 r} & -\frac{\left(F^2-1\right)\sigma_\theta\,\sigma_\phi}{F^2\,r^2} & 0 \end{pmatrix} \tag{\%t75}$$

Clean up

(%i79)
```
forget(0≤r)$
forget(0≤θ,θ≤π)$
forget(0≤sin(θ))$
forget(0≤ϕ,ϕ≤2*π)$
```

3 Bence Racskó wrote:

There is also an explicit procedure that is often better if the vielbein is simple.

We have

$$\mathrm{d}e^a = -\frac{1}{2}C^a_{bc}e^b \wedge e^c,$$

where the C^a_{bc} are the vielbein commutators. We can invert the first structure equation explicitly as

$$\begin{aligned} 0 = \mathrm{d}e^a + \omega^a{}_b \wedge e^b &= -\frac{1}{2}C^a_{bc}e^b \wedge e^c + \omega^a_{c\ b}e^c \wedge e^b \\ \frac{1}{2}C^a_{bc}e^b \wedge e^c &= \frac{1}{2}\left(\omega^a_{b\ c} - \omega^a_{c\ b}\right)e^b \wedge e^c, \end{aligned}$$

so

$$C^a_{bc} = \omega^a_{b\ c} - \omega^a_{c\ b}.$$

Lowering the index, we get

$$\begin{aligned} C_{a,bc} &= \omega_{b,ac} - \omega_{c,ab} \\ C_{b,ca} &= \omega_{c,ba} - \omega_{a,bc} \\ -C_{c,ab} &= -\omega_{a,cb} + \omega_{b,ca}, \end{aligned}$$

now sum these up:

$$\begin{aligned} C_{a,bc} + C_{b,ca} - C_{c,ab} &= 2\omega_{c,ba} \\ \omega_{c,ab} &= \frac{1}{2}\left(C_{c,ab} - C_{a,bc} - C_{b,ca}\right) \\ \omega_{ab} &= \frac{1}{2}\left(C_{c,ab} - C_{a,bc} - C_{b,ca}\right)e^c. \end{aligned}$$

If the veilbein is simple, then the $\mathrm{d}e^a = -\frac{1}{2}C^a_{bc}e^b \wedge e^c$ will only involve a few terms at most, and the spin connection is very easy to calculate from this.

4 JamalS wrote:

Looks like both our methods are relatively equal in terms of simplicity - if the vielbein is simple, as you said, in the first method solving the linear system is also trivial.

5 Bence Racskó wrote:

But I guess this is highly subjective. For vielbeins like the natural FLRW or Schwarzschild vielbeins, I find this method simpler and easier. But that's me.