Ansatz Schwarzschild metric

Based on Narcos Alpha Playlist PSI 18/19 - Gravitational Physics Review

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2)
```
info:build_info()$info@version;
```

(%o2)

5.38.1

(%i2)
```
reset()$kill(all)$
```

(%i1)
```
derivabbrev:true$
```

(%i2)
```
ratprint:false$
```

(%i3)
```
fpprintprec:5$
```

(%i4)
```
if get('vect,'version)=false then load(vect)$
```

(%i5)
```
if get('cartan,'version)=false then load(cartan)$
```

(%i6)
```
if get('format,'version)=false then load(format)$
```

(%i7)
```
declare(trigsimp,evfun)$
```

(%i11)
```
assume(0≤r)$
assume(0≤θ,θ≤π)$
assume(0≤sin(θ))$
assume(0≤ϕ,ϕ≤2*π)$
```

(%i12)
```
ξ:[t,r,θ,ϕ]$
```

(%i13)
```
dim:length(ξ)$
```

Line Element

(%i14)
```
depends([A,B,C],r)$
```

(%i15)
```
assume(A>0,B>0,C>0)$
```

(%i16)
```
ldisplay(ds^2=line_element:A^2*del(t)^2-B^2*del(r)^2-C^2*del(θ)^2-C^2*sin(θ)^2*del(ϕ)^2)$
```

$$ds^2 = -C^2 \sin(\theta)^2 \operatorname{del}(\phi)^2 - C^2 \operatorname{del}(\theta)^2 + A^2 \operatorname{del}(t)^2 - B^2 \operatorname{del}(r)^2 \quad (\%t16)$$

Covariant Metric Tensor

(%i20)
```
lg:zeromatrix(dim,dim)$
for i thru dim do
lg[i,i]:factor(coeff(expand(line_element),del(ξ[i])^2))$
for j thru dim do for k thru dim do
if j≠k then lg[j,k]:factor(expand(ratsimp(1/2*coeff(coeff(expand(line_element),del(ξ[j])),del(ξ[k]
ldisplay(lg)$
```

$$lg = \begin{pmatrix} A^2 & 0 & 0 & 0 \\ 0 & -B^2 & 0 & 0 \\ 0 & 0 & -C^2 & 0 \\ 0 & 0 & 0 & -C^2 \sin(\theta)^2 \end{pmatrix} \quad (\%t20)$$

Contravariant Metric Tensor

(%i21) `ldisplay(ug:invert(lg))$`

$$ug = \begin{pmatrix} \frac{1}{A^2} & 0 & 0 & 0 \\ 0 & -\frac{1}{B^2} & 0 & 0 \\ 0 & 0 & -\frac{1}{C^2} & 0 \\ 0 & 0 & 0 & -\frac{1}{C^2 \sin(\theta)^2} \end{pmatrix} \tag{\%t21}$$

Line element

(%i22) `ldisplay(ds`^{2}`=diff(`ξ`).lg.transpose(diff(`ξ`)))$`

$$ds^2 = -C^2 \sin(\theta)^2 \operatorname{del}(\phi)^2 - C^2 \operatorname{del}(\theta)^2 + A^2 \operatorname{del}(t)^2 - B^2 \operatorname{del}(r)^2 \tag{\%t22}$$

Define the frame e

(%i26)
```
e[t]:√(ug)[1]$
e[r]:√(-ug)[2]$
e[θ]:√(-ug)[3]$
e[ϕ]:√(-ug)[4]$
```

(%i27) `ldisplay(e:apply('matrix,[e[t],e[r],e[`θ`],e[`ϕ`]]))$`

$$e = \begin{pmatrix} \frac{1}{A} & 0 & 0 & 0 \\ 0 & \frac{1}{B} & 0 & 0 \\ 0 & 0 & \frac{1}{C} & 0 \\ 0 & 0 & 0 & \frac{1}{C \sin(\theta)} \end{pmatrix} \tag{\%t27}$$

Initialize cartan package

(%i28) `init_cartan(`ξ`)$`

(%i29) `cartan_basis;`

$$[dt, dr, d\theta, d\phi] \tag{\%o29}$$

(%i30) `cartan_coords;`

$$[t, r, \theta, \phi] \tag{\%o30}$$

(%i31) `cartan_dim;`

$$4 \tag{\%o31}$$

(%i32) `extdim;`

$$4 \tag{\%o32}$$

Define the coframe ω

(%i38)
```
kill(ω)$
ω[t]:list_matrix_entries(√(lg).cartan_basis)[1]$
ω[r]:list_matrix_entries(√(-lg).cartan_basis)[2]$
ω[θ]:list_matrix_entries(√(-lg).cartan_basis)[3]$
ω[φ]:list_matrix_entries(√(-lg).cartan_basis)[4]$
ldisplay(ω:[ω[t],ω[r],ω[θ],ω[φ]])$
```

$$\omega = [A\, dt, B\, dr, C\, d\theta, C\, d\phi\, \sin(\theta)] \tag{\%t38}$$

Verify $\langle \underline{\omega}^a \mid \underline{e}_b \rangle = \delta^a{}_b$

(%i39)
```
genmatrix(lambda([i,j],e[ξ[i]]|ω[ξ[j]]),cartan_dim,cartan_dim);
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix} \tag{\%o39}$$

Calculate the external derivative of the coframe $\mathrm{d}\omega$

(%i40)
```
ldisplay(dω:ext_diff(ω))$
```

$$d\omega = [-(A_r)\ dr\, dt, 0, (C_r)\ dr\, d\theta, (C_r)\ dr\, d\phi\, \sin(\theta) + C\, d\theta\, d\phi\, \cos(\theta)] \tag{\%t40}$$

Generic Connection 1-form Θ

(%i46)
```
A_a:[a_1,a_2,a_3,a_4]$
A_b:[b_1,b_2,b_3,b_4]$
A_c:[c_1,c_2,c_3,c_4]$
A_d:[d_1,d_2,d_3,d_4]$
A_e:[e_1,e_2,e_3,e_4]$
A_f:[f_1,f_2,f_3,f_4]$
```

(%i54)
```
kill(Θ)$
Θ:zeromatrix(dim,dim)$
Θ[1,2]:-Θ[2,1]:A_a.cartan_basis$
Θ[1,3]:-Θ[3,1]:A_b.cartan_basis$
Θ[1,4]:-Θ[4,1]:A_c.cartan_basis$
Θ[2,3]:-Θ[3,2]:A_d.cartan_basis$
Θ[2,4]:-Θ[4,2]:A_e.cartan_basis$
Θ[3,4]:-Θ[4,3]:A_f.cartan_basis$
```

Change matrix multiplication operator

(%i55)
```
matrix_element_mult:"~"$
```

$\lambda^a = \Theta_b{}^a \wedge \omega^b$

(%i56)
```
λ:list_matrix_entries(expand(Θ.ω))$
```

Restore matrix multiplication operator

(%i57) `matrix_element_mult:"*"$`

Cartan's First structural equation $\mathrm{d}\omega^{i} = \Theta_{j}{}^{i} \wedge \omega^{j}$

(%i58) `Eq:zeromatrix(dim,dim)$`

(%i59)
```
for i thru dim do for j thru dim do
Eq[i,j]:format(coeff(coeff(dω,cartan_basis[i]),cartan_basis[j])=
coeff(coeff(-λ,cartan_basis[i]),cartan_basis[j]),%list)$
```

(%i60) `Eqs:apply('append,list_matrix_entries(Eq))$`

(%i61) `linsol:linsolve(Eqs,append(A_a,A_b,A_c,A_d,A_e,A_f))$`

solve: dependent equations eliminated: (1 64 63 62 61 2 3 4 21 22 23 24 41 42 43 44 18 35 52 33 38 47 49 30 45 34 36 51 50

(%i62) `ldisplay(λ:at(λ,linsol))$`

$$\lambda = \left[(A_r)\ dr\ dt, 0, -(C_r)\ dr\ d\theta, -(C_r)\ dr\ d\phi\ \sin(\theta) - C\ d\theta\ d\phi\ \cos(\theta)\right] \qquad (\%t62)$$

(%i63) `is(dω=-λ);`

$$\text{true} \qquad (\%o63)$$

Update Connection 1-form Θ

(%i64) `ldisplay(Θ:at(Θ,linsol))$`

$$\Theta = \begin{pmatrix} 0 & \frac{(A_r)\ dt}{B} & 0 & 0 \\ -\frac{(A_r)\ dt}{B} & 0 & -\frac{(C_r)\ d\theta}{B} & -\frac{(C_r)\ d\phi\ \sin(\theta)}{B} \\ 0 & \frac{(C_r)\ d\theta}{B} & 0 & -d\phi\ \cos(\theta) \\ 0 & \frac{(C_r)\ d\phi\ \sin(\theta)}{B} & d\phi\ \cos(\theta) & 0 \end{pmatrix} \qquad (\%t64)$$

Exterior derivative of Connection 1-form $\mathrm{d}\Theta$

(%i65) `ldisplay(dΘ:fullratsimp(trigsimp(matrixmap(edit,ext_diff(Θ)))))$`

$$d\Theta = \left(\begin{array}{cccc} 0 & \frac{((A_r)\ (B_r)-(A_{rr})B)\ dr\ dt}{B^2} & 0 & \\ -\frac{((A_r)\ (B_r)-(A_{rr})B)\ dr\ dt}{B^2} & 0 & -\frac{(B\ (C_{rr})-(B_r)\ (C_r))\ dr\ d\theta}{B^2} & -\frac{d\phi\ ((B\ (C_{rr})-(B_r)}{} \\ 0 & \frac{(B\ (C_{rr})-(B_r)\ (C_r))\ dr\ d\theta}{B^2} & 0 & d \\ 0 & \frac{d\phi\ ((B\ (C_{rr})-(B_r)\ (C_r))\ dr\ \sin(\theta)+B\ (C_r)\ d\theta\ \cos(\theta))}{B^2} & -d\theta\ d\phi\ \sin(\theta) & \end{array} \right. \qquad (\%t65)$$

Update coefficients

(%i71)
```
ldisplay(A_a:at(A_a,linsol))$
ldisplay(A_b:at(A_b,linsol))$
ldisplay(A_c:at(A_c,linsol))$
ldisplay(A_d:at(A_d,linsol))$
ldisplay(A_e:at(A_e,linsol))$
ldisplay(A_f:at(A_f,linsol))$
```

$$A_a = \left[-\frac{A_r}{B}, 0, 0, 0\right] \qquad (\%t66)$$

$$A_b = [0, 0, 0, 0] \tag{\%t67}$$

$$A_c = [0, 0, 0, 0] \tag{\%t68}$$

$$A_d = \left[0, 0, \frac{C_r}{B}, 0\right] \tag{\%t69}$$

$$A_e = \left[0, 0, 0, \frac{(C_r)\ \sin(\theta)}{B}\right] \tag{\%t70}$$

$$A_f = [0, 0, 0, \cos(\theta)] \tag{\%t71}$$

Change matrix multiplication operator

(%i72) `matrix_element_mult:"~"$`

Cartan's Second structural equation: $\Omega_j{}^i = \mathrm{d}\Theta_j{}^i + \Theta_k{}^i \wedge \Theta_j{}^k$

Curvature 2-form Ω

(%i73) `ldisplay(Ω:fullratsimp(matrixmap(edit,dΘ+Θ.Θ)))$`

$$\Omega = \begin{pmatrix} 0 & \frac{((A_r)\,(B_r)-(A_{rr})B)\ dr\ dt}{B^2} & -\frac{(A_r)\,(C_r)\ dt\ d\theta}{B^2} & -\frac{(A_r)\,(C_r)\ dt\ d\phi\ \sin(\theta)}{B^2} \\ -\frac{((A_r)\,(B_r)-(A_{rr})B)\ dr\ dt}{B^2} & 0 & -\frac{(B\,(C_{rr})-(B_r)\,(C_r))\ dr\ d\theta}{B^2} & -\frac{(B\,(C_{rr})-(B_r)\,(C_r))\ dr\ d\phi\ \sin(\theta)}{B^2} \\ \frac{(A_r)\,(C_r)\ dt\ d\theta}{B^2} & \frac{(B\,(C_{rr})-(B_r)\,(C_r))\ dr\ d\theta}{B^2} & 0 & -\frac{\left((C_r)^2-B^2\right)\ d\theta\ d\phi\ \sin(\theta)}{B^2} \\ \frac{(A_r)\,(C_r)\ dt\ d\phi\ \sin(\theta)}{B^2} & \frac{(B\,(C_{rr})-(B_r)\,(C_r))\ dr\ d\phi\ \sin(\theta)}{B^2} & \frac{\left((C_r)^2-B^2\right)\ d\theta\ d\phi\ \sin(\theta)}{B^2} & 0 \end{pmatrix} \tag{\%t73}$$

(%i74)
```
for i thru dim do for j:i thru dim do
if Ω[i,j]≠0 then ldisplay('Ω[ξ[i],ξ[j]]=fullratsimp(Ω[i,j]))$
```

$$\Omega_{t,r} = \frac{((A_r)\ (B_r) - (A_{rr})\,B)\ dr\ dt}{B^2} \tag{\%t74}$$

$$\Omega_{t,\theta} = -\frac{(A_r)\ (C_r)\ dt\ d\theta}{B^2} \tag{\%t75}$$

$$\Omega_{t,\phi} = -\frac{(A_r)\ (C_r)\ dt\ d\phi\ \sin(\theta)}{B^2} \tag{\%t76}$$

$$\Omega_{r,\theta} = -\frac{(B\ (C_{rr}) - (B_r)\ (C_r))\ dr\ d\theta}{B^2} \tag{\%t77}$$

$$\Omega_{r,\phi} = -\frac{(B\ (C_{rr}) - (B_r)\ (C_r))\ dr\ d\phi\ \sin(\theta)}{B^2} \tag{\%t78}$$

$$\Omega_{\theta,\phi} = -\frac{\left((C_r)^2 - B^2\right)\ d\theta\ d\phi\ \sin(\theta)}{B^2} \tag{\%t79}$$

Restore matrix multiplication operator

(%i80) `matrix_element_mult:"*"$`

Riemann tensor R

(%i83)
```
kill(R)$
array(R,dim,dim,dim,dim)$
for α thru dim do for β thru dim do
for γ thru dim do for δ thru dim do
R[α,β,γ,δ]:e[ξ[δ]]|(e[ξ[γ]]|Ω[α,β])$
```

(%i84)
```
for α thru dim do for β:α thru dim do
for γ thru α do for δ thru β do
if R[α,β,γ,δ]≠0 then
ldisplay('R[ξ[α],ξ[β],ξ[γ],ξ[δ]]=fullratsimp(R[α,β,γ,δ]))$
```

$$R_{t,r,t,r} = \frac{(A_r)\ (B_r) - (A_{rr})\, B}{A\, B^3} \tag{\%t84}$$

$$R_{t,\theta,t,\theta} = -\frac{(A_r)\ (C_r)}{A\, B^2 C} \tag{\%t85}$$

$$R_{t,\phi,t,\phi} = -\frac{(A_r)\ (C_r)}{A\, B^2 C} \tag{\%t86}$$

$$R_{r,\theta,r,\theta} = -\frac{B\ (C_{rr}) - (B_r)\ (C_r)}{B^3 C} \tag{\%t87}$$

$$R_{r,\phi,r,\phi} = -\frac{B\ (C_{rr}) - (B_r)\ (C_r)}{B^3 C} \tag{\%t88}$$

$$R_{\theta,\phi,\theta,\phi} = -\frac{(C_r)^2 - B^2}{B^2\, C^2} \tag{\%t89}$$

Forms in terms of the coframe σ

(%i90)
```
kill(labels)$
```

(%i1)
```
Eqs:makelist(σ[ξ[i]]=ω[ξ[i]],i,1,cartan_dim);
```

$$[\sigma_t = A\, dt, \sigma_r = B\, dr, \sigma_\theta = C\, d\theta, \sigma_\phi = C\, d\phi\, \sin(\theta)] \tag{Eqs}$$

(%i2)
```
linsol:linsolve(Eqs,cartan_basis);
```

$$\left[dt = \frac{\sigma_t}{A}, dr = \frac{\sigma_r}{B}, d\theta = \frac{\sigma_\theta}{C}, d\phi = \frac{\sigma_\phi}{C\, \sin(\theta)}\right] \tag{linsol}$$

Connection 1-form Θ

(%i3)
```
ldisplay(Θ:ev(Θ,linsol,fullratsimp))$
```

$$\Theta = \begin{pmatrix} 0 & \frac{(A_r)\,\sigma_t}{AB} & 0 & 0 \\ -\frac{(A_r)\,\sigma_t}{AB} & 0 & -\frac{(C_r)\,\sigma_\theta}{BC} & -\frac{(C_r)\,\sigma_\phi}{BC} \\ 0 & \frac{(C_r)\,\sigma_\theta}{BC} & 0 & -\frac{\cos(\theta)\,\sigma_\phi}{C\,\sin(\theta)} \\ 0 & \frac{(C_r)\,\sigma_\phi}{BC} & \frac{\cos(\theta)\,\sigma_\phi}{C\,\sin(\theta)} & 0 \end{pmatrix} \tag{\%t3}$$

Curvature 2-form Ω

(%i5)
```
Ω:ev(Ω,linsol,fullratsimp)$
ldisplay(Ω)$
```

$$\Omega = \begin{pmatrix} 0 & \frac{((A_r)\,(B_r)-(A_{rr})B)\,\sigma_r\,\sigma_t}{A\,B^3} & -\frac{(A_r)\,(C_r)\,\sigma_t\,\sigma_\theta}{A\,B^2 C} & -\frac{(A_r)\,(C_r)\,\sigma_t\,\sigma_\phi}{A\,B^2 C} \\ -\frac{((A_r)\,(B_r)-(A_{rr})B)\,\sigma_r\,\sigma_t}{A\,B^3} & 0 & -\frac{(B\,(C_{rr})-(B_r)\,(C_r))\,\sigma_r\,\sigma_\theta}{B^3 C} & -\frac{(B\,(C_{rr})-(B_r)\,(C_r))\,\sigma_r\,\sigma_\phi}{B^3 C} \\ \frac{(A_r)\,(C_r)\,\sigma_t\,\sigma_\theta}{A\,B^2 C} & \frac{(B\,(C_{rr})-(B_r)\,(C_r))\,\sigma_r\,\sigma_\theta}{B^3 C} & 0 & -\frac{\left((C_r)^2-B^2\right)\sigma_\theta\,\sigma_\phi}{B^2\,C^2} \\ \frac{(A_r)\,(C_r)\,\sigma_t\,\sigma_\phi}{A\,B^2 C} & \frac{(B\,(C_{rr})-(B_r)\,(C_r))\,\sigma_r\,\sigma_\phi}{B^3 C} & \frac{\left((C_r)^2-B^2\right)\sigma_\theta\,\sigma_\phi}{B^2\,C^2} & 0 \end{pmatrix} \tag{\%t5}$$

(%i6)
```
for i thru dim do for j:i thru dim do
if Ω[i,j]≠0 then ldisplay('Ω[ξ[i],ξ[j]]=fullratsimp(Ω[i,j]))$
```

$$\Omega_{t,r} = \frac{((A_r)\;(B_r) - (A_{rr})\;B)\;\sigma_r\,\sigma_t}{A\,B^3} \tag{\%t6}$$

$$\Omega_{t,\theta} = -\frac{(A_r)\;(C_r)\;\sigma_t\,\sigma_\theta}{A\,B^2 C} \tag{\%t7}$$

$$\Omega_{t,\phi} = -\frac{(A_r)\;(C_r)\;\sigma_t\,\sigma_\phi}{A\,B^2 C} \tag{\%t8}$$

$$\Omega_{r,\theta} = -\frac{(B\;(C_{rr}) - (B_r)\;(C_r))\;\sigma_r\,\sigma_\theta}{B^3 C} \tag{\%t9}$$

$$\Omega_{r,\phi} = -\frac{(B\;(C_{rr}) - (B_r)\;(C_r))\;\sigma_r\,\sigma_\phi}{B^3 C} \tag{\%t10}$$

$$\Omega_{\theta,\phi} = -\frac{\left((C_r)^2 - B^2\right)\;\sigma_\theta\,\sigma_\phi}{B^2\,C^2} \tag{\%t11}$$

Schwarzschild gauge

(%i14)
```
A:√((r-2*m)/r)$
B:√(r/(r-2*m))$
C:r$
```

Line Element

(%i15)
```
ldisplay(ds²=line_element:ev(line_element))$
```

$$ds^2 = -r^2\sin(\theta)^2\,\mathrm{del}(\phi)^2 - r^2\,\mathrm{del}(\theta)^2 + \frac{(r-2m)\;\mathrm{del}(t)^2}{r} - \frac{r\,\mathrm{del}(r)^2}{r-2m} \tag{\%t15}$$

Connection 1-form Θ

(%i16)
```
ldisplay(Θ:ev(Θ,diff,eval,fullratsimp))$
```

$$\Theta = \begin{pmatrix} 0 & \frac{m\,\sigma_t}{r^{\frac{3}{2}}\,\sqrt{r-2m}} & 0 & 0 \\ -\frac{m\,\sigma_t}{r^{\frac{3}{2}}\,\sqrt{r-2m}} & 0 & -\frac{\sqrt{r-2m}\,\sigma_\theta}{r^{\frac{3}{2}}} & -\frac{\sqrt{r-2m}\,\sigma_\phi}{r^{\frac{3}{2}}} \\ 0 & \frac{\sqrt{r-2m}\,\sigma_\theta}{r^{\frac{3}{2}}} & 0 & -\frac{\cos(\theta)\,\sigma_\phi}{r\,\sin(\theta)} \\ 0 & \frac{\sqrt{r-2m}\,\sigma_\phi}{r^{\frac{3}{2}}} & \frac{\cos(\theta)\,\sigma_\phi}{r\,\sin(\theta)} & 0 \end{pmatrix} \tag{\%t16}$$

Curvature 2-form Ω

(%i17) `ldisplay(Ω:ev(Ω,diff,eval,fullratsimp))$`

$$\Omega = \begin{pmatrix} 0 & \frac{2m\,\sigma_r\,\sigma_t}{r^3} & -\frac{m\,\sigma_t\,\sigma_\theta}{r^3} & -\frac{m\,\sigma_t\,\sigma_\phi}{r^3} \\ -\frac{2m\,\sigma_r\,\sigma_t}{r^3} & 0 & -\frac{m\,\sigma_r\,\sigma_\theta}{r^3} & -\frac{m\,\sigma_r\,\sigma_\phi}{r^3} \\ \frac{m\,\sigma_t\,\sigma_\theta}{r^3} & \frac{m\,\sigma_r\,\sigma_\theta}{r^3} & 0 & \frac{2m\,\sigma_\theta\,\sigma_\phi}{r^3} \\ \frac{m\,\sigma_t\,\sigma_\phi}{r^3} & \frac{m\,\sigma_r\,\sigma_\phi}{r^3} & -\frac{2m\,\sigma_\theta\,\sigma_\phi}{r^3} & 0 \end{pmatrix} \tag{\%t17}$$

Clean up

(%i21)
```
forget(0≤r)$
forget(0≤θ,θ≤π)$
forget(0≤sin(θ))$
forget(0≤φ,φ≤2*π)$
```

(%i22) `elapsed_real_time();`

$$15.346 \tag{\%o22}$$

(%i23) `elapsed_run_time();`

$$1.766 \tag{\%o23}$$