Flux across a hemisphere

Based on Dr. Bevin Maultsby Playlist Flux across a hemisphere, with and without the Divergence Theorem

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2) `info:build_info()$info@version;`

(%o2)

5.38.1

(%i2) `reset()$kill(all)$`

(%i1) `derivabbrev:true$`

(%i2) `ratprint:false$`

(%i3) `fpprintprec:5$`

(%i4) `load(linearalgebra)$`

(%i5) `if get('draw,'version)=false then load(draw)$`

(%i6) `wxplot_size:[1024,768]$`

(%i7)
```
set_draw_defaults(xtics=1,ytics=1,ztics=1,xyplane=0,nticks=100,
xaxis=true,xaxis_type=dots,xaxis_width=3,
yaxis=true,yaxis_type=dots,yaxis_width=3,
zaxis=true,zaxis_type=dots,zaxis_width=3,
background_color=light_gray)$
```

(%i8) `if get('vect,'version)=false then load(vect)$`

(%i9) `norm(u):=block(ratsimp(radcan(√(u.u))))$`

(%i10) `normalize(v):=block(v/norm(v))$`

(%i11) `angle(u,v):=block([junk:radcan(√((u.u)*(v.v)))],acos(u.v/junk))$`

(%i12) `mycross(va,vb):=[va[2]*vb[3]-va[3]*vb[2],va[3]*vb[1]-va[1]*vb[3],va[1]*vb[2]-va[2]*vb[1]]$`

(%i13) `if get('cartan,'version)=false then load(cartan)$`

(%i14) `declare(trigsimp,evfun)$`

Let M be the surface $x^2 + y^2 + z^2 = 9$. Using the outward-pointing normal, find the flux through M for the vector field $\vec{F}(x, y, z) = \langle y, x, z \rangle$.

Define the space $\mathbb{R}^3$

(%i15)
```
ζ:[x,y,z]$
```

(%i16)
```
dim:length(ζ)$
```

(%i17)
```
scalefactors(ζ)$
```

(%i18)
```
init_cartan(ζ)$
```

Vector field $\vec{F} \in \mathbb{R}^3$

(%i19)
```
ldisplay(F:[y,x,z])$
```

$$F = [y, x, z] \tag{\%t19}$$

3D Direction field

(%i21)
```
/* vector origins are (x,y,z)| x,y=1,...,5 */
coord:setify(makelist(k,k,-3,3))$
points3d:listify(cartesian_product(coord,coord,coord))$
```

(%i23)
```
/* compute vectors at the given points */
define(vf3d(x,y,z),vector(ζ,F))$
vect3:makelist(vf3d(k[1],k[2],k[3]),k,points3d)$
```

(%i24) `wxdraw3d([head_length=0.1,color=blue,head_angle=25,unit_vectors=true],vect3)$`



(%t24)

Calculate $\nabla \times \vec{F} \in \mathbb{R}^3$

(%i25)
```
ldisplay(curlF:ev(express(curl(F)),diff))$
```

$$curlF = [0,0,0] \tag{%t25}$$

Work form $\alpha = F^{\flat} \in \mathcal{A}^1(\mathbb{R}^3)$

(%i26)
```
ldisplay(α:F.cartan_basis)$
```

$$\alpha = z\,dz + x\,dy + y\,dx \tag{%t26}$$

Calculate $\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

(%i27)
```
ldisplay(dα:ext_diff(α))$
```

$$d\alpha = 0 \tag{%t27}$$

Calculate $\nabla \cdot \vec{F} \in \mathbb{R}$

(%i28)
```
ldisplay(divF:ev(express(div(F)),diff))$
```

$$divF = 1 \tag{%t28}$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i29)
```
ldisplay(β:F[1]*cartan_basis[2]~cartan_basis[3]+
F[2]*cartan_basis[3]~cartan_basis[1]+
F[3]*cartan_basis[1]~cartan_basis[2])$
```

$$\beta = y\,dy\,dz - x\,dx\,dz + z\,dx\,dy \tag{%t29}$$

(%i30)
```
ϵ[i,j,k]:=1/2*(i-j)*(j-k)*(k-i)$
```

(%i31)
```
ldisplay(p:edit(1/2*sum(sum(sum(ϵ[i,j,k]*
F[i]*cartan_basis[j]~cartan_basis[k],
i,1,dim),j,1,dim),k,1,dim)))$
```

$$p = y\,dy\,dz - x\,dx\,dz + z\,dx\,dy \tag{%t31}$$

(%i32)
```
is(p=β);
```

$$\text{true} \tag{%o32}$$

Calculate $\mathrm{d}\beta \in \mathcal{A}^3(\mathbb{R}^3)$

(%i33)
```
ldisplay(dβ:ext_diff(β))$
```

$$d\beta = dx\,dy\,dz \tag{%t33}$$

Surface $\vec{S} \in \mathbb{R}^3$

(%i34) ldisplay(S:3*[cos(u)*sin(v),sin(u)*sin(v),cos(v)])$

$$S = [3\cos(u)\ \sin(v), 3\sin(u)\ \sin(v), 3\cos(v)] \qquad (\%t34)$$

(%i35) wxdraw3d(title="Surface",
xu_grid=100,yv_grid=100,view=[60,30],
proportional_axes=xyz,surface_hide=true,
color=green,
apply(parametric_surface,append(S,[u,0,2*π,v,0,$\frac{1}{2}$*π])))$



(%t35)

(%i36) `ldisplay(S_u:diff(S,u))$`

$$S_u = [-3\sin(u)\ \sin(v), 3\cos(u)\ \sin(v), 0] \quad (\%t36)$$

(%i37) `ldisplay(S_v:diff(S,v))$`

$$S_v = [3\cos(u)\ \cos(v), 3\sin(u)\ \cos(v), -3\sin(v)] \quad (\%t37)$$

Normal $n_S \in \mathbb{R}^3$

(%i38) `ldisplay(n_S:trigsimp(mycross(S_v,S_u)))$`

$$n_S = [9\cos(u)\sin(v)^2, 9\sin(u)\sin(v)^2, 9\cos(v)\ \sin(v)] \quad (\%t38)$$

(%i39) `is(n_S=3*sin(v)*S);`

$$\text{true} \quad (\%o39)$$

Calculate $\vec{F} \circ \vec{S}$

(%i40) `ldisplay(FoS:subst(map("=",ζ,S),F))$`

$$FoS = [3\sin(u)\ \sin(v), 3\cos(u)\ \sin(v), 3\cos(v)] \quad (\%t40)$$

Calculate $\alpha \circ \vec{S}$

(%i41) `ldisplay(αoS:subst(map("=",ζ,S),α))$`

$$\alpha oS = 3\cos(v)\ dz + 3\cos(u)\ \sin(v)\ dy + 3\sin(u)\ \sin(v)\ dx \quad (\%t41)$$

Integrand

(%i42) `integrand:trigsimp(n_S|αoS);`

$$(54\cos(u)\ \sin(u) - 27)\ \sin(v)^3 + 27\sin(v) \quad (\text{integrand})$$

(%i43) `integrand:trigsimp(FoS.n_S);`

$$(54\cos(u)\ \sin(u) - 27)\ \sin(v)^3 + 27\sin(v) \quad (\text{integrand})$$

Flux integral

(%i44) `I:'integrate('integrate(integrand,v,0,` $\frac{1}{2}$`*π),u,0,2*π)$`

(%i45) `ldisplay(I=box(ev(I,integrate)))$`

$$\int_0^{2\pi}\int_0^{\frac{\pi}{2}} (54\cos(u)\ \sin(u) - 27)\ \sin(v)^3 + 27\sin(v) dv\, du = \boxed{18\pi} \quad (\%t45)$$

Using the Divergence theorem

$$\iiint_E \nabla\cdot\vec{F}\,\mathrm{d}V = \iint_{\partial E}\vec{F}\cdot\mathrm{d}\vec{S} = \iint_S \vec{F}\cdot\mathrm{d}\vec{S} + \iint_L \vec{F}\cdot\mathrm{d}\vec{S}$$

Surface $\vec{L}\in\mathbb{R}^3$

(%i46) `ldisplay(L:[ρ*cos(θ),ρ*sin(θ),0])$`

$$L = [\cos(\theta)\rho, \sin(\theta)\rho, 0] \tag{\%t46}$$

(%i47)
```
wxdraw3d(title="Surface",
xu_grid=100,yv_grid=100,view=[60,30],
proportional_axes=xyz,surface_hide=false,
color=green,
apply(parametric_surface,append(S,[u,0,2*π,v,0,1/2*π])),
color=red,line_width=5,
apply(parametric_surface,append(L,[ρ,0,3,θ,0,2*π])))$
```



(%t47)

(%i48) `ldisplay(L_ρ:diff(L,ρ))$`

$$L_\rho = [\cos(\theta), \sin(\theta), 0] \tag{\%t48}$$

(%i49) `ldisplay(L_θ:diff(L,θ))$`

$$L_\theta = [-\sin(\theta)\rho, \cos(\theta)\rho, 0] \tag{\%t49}$$

Normal $n_L \in \mathbb{R}^3$

(%i50) `ldisplay(n_L:trigsimp(mycross(L_θ,L_ρ)))$`

$$n_L = [0, 0, -\rho] \tag{\%t50}$$

Calculate $\vec{F} \circ \vec{L}$

(%i51) `ldisplay(FoL:subst(map("=",ζ,L),F))$`

$$FoL = [\sin(\theta)\rho, \cos(\theta)\rho, 0] \tag{\%t51}$$

Calculate $\alpha \circ \vec{L}$

(%i52) `ldisplay(αoL:subst(map("=",ζ,L),α))$`

$$\alpha oL = dx\ \sin(\theta)\rho + dy\ \cos(\theta)\rho \tag{\%t52}$$

Integrand

(%i53) `integrand:trigsimp(n_L|αoL);`

$$0 \tag{integrand}$$

(%i54) `integrand:trigsimp(FoL.n_L);`

$$0 \tag{integrand}$$

Spherical coordinates

```
(%i58) assume(0≤ρ)$
       assume(0≤θ,θ≤π)$
       assume(0≤sin(θ))$
       assume(0≤ϕ,ϕ≤2*π)$
(%i59) ξ:[ρ,θ,ϕ]$
(%i60) ldisplay(E:[ρ*sin(θ)*cos(ϕ),ρ*sin(θ)*sin(ϕ),ρ*cos(θ)])$
```

$$E = [\sin(\theta)\rho\,\cos(\phi), \sin(\theta)\rho\,\sin(\phi), \cos(\theta)\rho] \tag{\%t60}$$

```
(%i61) scalefactors(append([E],ξ))$
(%i62) sf;
```

$$[1, \rho, \sin(\theta)\rho] \tag{\%o62}$$

```
(%i63) sfprod;
```

$$\sin(\theta)\,\rho^2 \tag{\%o63}$$

```
(%i64) dimension;
```

$$3 \tag{\%o64}$$

```
(%i65) ldisplay(J:trigsimp(determinant(jacobian(E,ξ))))$
```

$$J = \sin(\theta)\,\rho^2 \tag{\%t65}$$

```
(%i66) trigsimp(diff(E,ϕ)|(diff(E,θ)|(diff(E,ρ)|dβ)));
```

$$\sin(\theta)\,\rho^2 \tag{\%o66}$$

Calculate

$$\iiint_E \nabla\cdot\vec{F}\,\mathrm{d}V$$

```
(%i67) I:1/2*'integrate('integrate('integrate(divF*J,ρ,0,3),θ,0,π),ϕ,0,2*π)$
(%i68) ldisplay(I=box(ev(I,integrate)))$
```

$$\pi\int_0^{\pi}\sin(\theta)d\theta\,\int_0^3\rho^2 d\rho = (18\pi) \tag{\%t68}$$