Web REDUCE (6745), 2-Apr-2024 ...

```
% Based on Narcos Alpha Playlist
% https://www.youtube.com/playlist?list=PLhS8YPbZkfgIxU3dwrqsj30E9FJqNAMOj
% PSI 18/19 - Gravitational Physics Review
% Lectures by Ruth Gregory

load_package excalc$

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
write "Define Polar coordinates coframe"$
coframe o(r)     = 1 * d r,
        o(theta) = r * d theta
  with metric g = o(r) * o(r) + o(theta) * o(theta)$

frame e$
displayframe;
DETM!*;

on fancy;
on nero$
factor o,^$

write "Verify"$
e(-k) _| o(l);

clear omega$
riemannconx omega$
write "Display the connection form"$
omega(k,-l);

write "Display the connection form in Matrix"$
coords := {r, theta}$
matrix Momega(2, 2)$
for k := 1:2 do for l := 1:2 do
Momega(k, l) := omega(part(coords, k), part(coords, l))$
Momega;

clear curv,riemann,ricci,riccisc$
pform curv(k,l)=2,{riemann(a,b,c,d),ricci(a,b),riccisc}=0$

index_symmetries curv(k,l): antisymmetric,
                 riemann(k,l,m,n): antisymmetric in {k,l},{m,n}
                                   symmetric in {{k,l},{m,n}},
                 ricci(k,l): symmetric;

write "Display the curvature form"$
curv(k,-l) := d omega(k,-l) + omega(k,-m) ^ omega(m,-l);

write "Display the curvature form in Matrix"$
matrix Mcurv(2, 2)$
for k := 1:2 do for l := 1:2 do
Mcurv(k, l) := curv(part(coords, k), part(coords, l))$
Mcurv;

write "Display the Riemann Tensor"$
riemann(a,b,c,d) := e(d) _| ( e(c) _| curv(a,b) );

write "Display the Ricci Tensor"$
ricci(-a,-b) := riemann(c,-a,-d,-b) * g(-c,d);
```

```
write "Display the Ricci Scalar"$
riccisc := ricci(-a,-b) * g(a,b);

write "Display the Einstein Tensor"$
clear einstein$
pform einstein(a)=3$
einstein(-a) := (1/2) * curv(b,c) ^ #( o(-b) ^ o(-c) ^ o(-a) );

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
write "Define the 2-Sphere coframe"$
coframe o(theta)  =          1 * d theta,
        o(phi)    = sin(theta) * d phi
  with metric g = o(theta) * o(theta) + o(phi) * o(phi)$

frame e$
displayframe;
DETM!*;

on fancy;
on nero$
factor o,^$

write "Verify"$
e(-k) _| o(l);

clear omega$
riemannconx omega$
write "Display the connection form"$
omega(k,-l);

write "Display the connection form in Matrix"$
coords := {theta, phi}$
matrix Momega(2, 2)$
for k := 1:2 do for l := 1:2 do
Momega(k, l) := omega(part(coords, k), part(coords, l))$
Momega;

clear curv,riemann,ricci,riccisc$
pform curv(k,l)=2,{riemann(a,b,c,d),ricci(a,b),riccisc}=0$

index_symmetries curv(k,l): antisymmetric,
                 riemann(k,l,m,n): antisymmetric in {k,l},{m,n}
                                   symmetric in {{k,l},{m,n}},
                 ricci(k,l): symmetric;

write "Display the curvature form"$
curv(k,-l) := d omega(k,-l) + omega(k,-m) ^ omega(m,-l);

write "Display the curvature form in Matrix"$
matrix Mcurv(2, 2)$
for k := 1:2 do for l := 1:2 do
Mcurv(k, l) := curv(part(coords, k), part(coords, l))$
Mcurv;

write "Display the Riemann Tensor"$
riemann(a,b,c,d) := e(d) _| ( e(c) _| curv(a,b) );

write "Display the Ricci Tensor"$
ricci(-a,-b) := riemann(c,-a,-d,-b) * g(-c,d);
```

```
write "Display the Ricci Scalar"$
riccisc := ricci(-a,-b) * g(a,b);

write "Display the Einstein Tensor"$
clear einstein$
pform einstein(a)=3$
einstein(-a) := (1/2) * curv(b,c) ^ #( o(-b) ^ o(-c) ^ o(-a) );

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
write "Define Spherical coordinates coframe"$
coframe o(r)     = 1              * d r,
        o(theta) = r              * d theta,
        o(phi)   = r * sin(theta) * d phi
  with metric g = o(r) * o(r) + o(theta) * o(theta) + o(phi) * o(phi)$

frame e$
displayframe;
DETM!*;

on fancy;
on nero$
factor o,^$

write "Verify"$
e(-k) _| o(l);

clear omega$
riemannconx omega$
write "Display the connection form"$
omega(k,-l);

write "Display the connection form in Matrix"$
coords := {r, theta, phi}$
matrix Momega(3, 3)$
for k := 1:3 do for l := 1:3 do
Momega(k, l) := omega(part(coords, k), part(coords, l))$
Momega;

clear curv,riemann,ricci,riccisc$
pform curv(k,l)=2,{riemann(a,b,c,d),ricci(a,b),riccisc}=0$

index_symmetries curv(k,l): antisymmetric,
                 riemann(k,l,m,n): antisymmetric in {k,l},{m,n}
                                   symmetric in {{k,l},{m,n}},
                 ricci(k,l): symmetric;

write "Display the curvature form"$
curv(k,-l) := d omega(k,-l) + omega(k,-m) ^ omega(m,-l);

write "Display the curvature form in Matrix"$
matrix Mcurv(3, 3)$
for k := 1:3 do for l := 1:3 do
Mcurv(k, l) := curv(part(coords, k), part(coords, l))$
Mcurv;

write "Display the Riemann Tensor"$
riemann(a,b,c,d) := e(d) _| ( e(c) _| curv(a,b) );

write "Display the Ricci Tensor"$
ricci(-a,-b) := riemann(c,-a,-d,-b) * g(-c,d);
```

```
write "Display the Ricci Scalar"$
riccisc := ricci(-a,-b) * g(a,b);

write "Display the Einstein Tensor"$
clear einstein$
pform einstein(a)=3$
einstein(-a) := (1/2) * curv(b,c) ^ #( o(-b) ^ o(-c) ^ o(-a) );

showtime;
end;
```

```
*** .*. redefined

*** × redefined

*** ^ redefined
```

Define Polar coordinates coframe

$$o^{r} = d\,r$$
$$o^{\theta} = d\,\theta\,r$$

1

Verify

$$\mathrm{ns}_{r}{}^{r} := 1$$
$$\mathrm{ns}_{\theta}{}^{\theta} := 1$$

Display the connection form

$$\omega^{\theta}{}_{r} := \frac{o^{\theta}}{r}$$
$$\omega^{r}{}_{\theta} := \frac{-o^{\theta}}{r}$$

Display the connection form in Matrix

$$\begin{pmatrix} 0 & \frac{-o^{\theta}}{r} \\ \frac{o^{\theta}}{r} & 0 \end{pmatrix}$$

$$\text{Display the curvature form}$$

$$\text{Display the curvature form in Matrix}$$

$$\begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}$$

$$\text{Display the Riemann Tensor}$$

$$\text{Display the Ricci Tensor}$$

$$\text{Display the Ricci Scalar}$$

$$\text{Display the Einstein Tensor}$$

$$\text{Define the } 2 - \text{Sphere coframe}$$

$$o^{\theta} = d\,\theta$$
$$o^{\phi} = d\,\phi\,\sin(\theta)$$

1

Verify

$$\mathrm{ns}_{\theta}{}^{\theta} := 1$$
$$\mathrm{ns}_{\phi}{}^{\phi} := 1$$

Display the connection form

$$\omega^{\phi}{}_{\theta} := \frac{o^{\phi}\,\cos(\theta)}{\sin(\theta)}$$
$$\omega^{\theta}{}_{\phi} := \frac{-o^{\phi}\,\cos(\theta)}{\sin(\theta)}$$

Display the connection form in Matrix

$$\begin{pmatrix} 0 & \frac{-o^{\phi}\,\cos(\theta)}{\sin(\theta)} \\ \frac{o^{\phi}\,\cos(\theta)}{\sin(\theta)} & 0 \end{pmatrix}$$

Display the curvature form

$$\mathrm{curv}^{\phi}{}_{\theta} := -\,o^{\theta} \wedge o^{\phi}$$
$$\mathrm{curv}^{\theta}{}_{\phi} := o^{\theta} \wedge o^{\phi}$$

Display the curvature form in Matrix

$$\begin{pmatrix} 0 & o^{\theta} \wedge o^{\phi} \\ -o^{\theta} \wedge o^{\phi} & 0 \end{pmatrix}$$

Display the Riemann Tensor

$$\text{riemann}^{\theta\,\phi\,\theta\,\phi}{:=}1$$

Display the Ricci Tensor

$$\text{ricci}_{\theta\,\theta}{:=}1$$

$$\text{ricci}_{\phi\,\phi}{:=}1$$

Display the Ricci Scalar

$$\text{riccisc}{:=}2$$

Display the Einstein Tensor

Define Spherical coordinates coframe

$$\begin{aligned} o^{r} &= d\,r \\ o^{\theta} &= d\,\theta\,r \\ o^{\phi} &= d\,\phi\,\sin\left(\theta\right)r \end{aligned}$$

Verify

$$\mathrm{ns}_{r}{}^{r} := 1$$

$$\mathrm{ns}_{\theta}{}^{\theta} := 1$$

$$\mathrm{ns}_{\phi}{}^{\phi} := 1$$

Display the connection form

$$\omega^{\theta}{}_{r} := \frac{o^{\theta}}{r}$$

$$\omega^{\phi}{}_{r} := \frac{o^{\phi}}{r}$$

$$\omega^{r}{}_{\theta} := \frac{-o^{\theta}}{r}$$

$$\omega^{\phi}{}_{\theta} := \frac{o^{\phi}\cos(\theta)}{\sin(\theta)\, r}$$

$$\omega^{r}{}_{\phi} := \frac{-o^{\phi}}{r}$$

$$\omega^{\theta}{}_{\phi} := \frac{-o^{\phi}\cos(\theta)}{\sin(\theta)\, r}$$

Display the connection form in Matrix

$$\begin{pmatrix} 0 & \frac{-o^{\theta}}{r} & \frac{-o^{\phi}}{r} \\ \frac{o^{\theta}}{r} & 0 & \frac{-o^{\phi}\cos(\theta)}{\sin(\theta)\, r} \\ \frac{o^{\phi}}{r} & \frac{o^{\phi}\cos(\theta)}{\sin(\theta)\, r} & 0 \end{pmatrix}$$

Display the curvature form

Display the curvature form in Matrix

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

Display the Riemann Tensor

Display the Ricci Tensor

Display the Ricci Scalar

Display the Einstein Tensor

```
Time: 25970 ms  plus GC time: 5 ms
```